
Učebné texty – Algoritmizácia a programovanie

1.1. Základné pojmy

Algoritmus je postup riešenia nejakého problému (postupnosť príkazov, ktoré procesor dokáže zrealizovať a ktoré vedú k vyriešeniu daného problému).

Algoritmus je postup transformácie vstupných dát na výstupné (vstupné dáta sú tie, ktoré procesoru na začiatku zadávame, výstupné sú výsledky, ktoré vypočíta.)


Vlastnosti algoritmu:
1. hromadnosť - algoritmus rieši nie jeden konkrétny problém, ale celú skupinu úloh podobného charakteru. Napr. nemá význam vyrábať algoritmus pre riešenie konkrétnej kvadratickej rovnice x2+5x+6=0, ale môžeme vyrobiť algoritmus pre riešenie kvadratickej rovnice s ľubovoľnými koeficientami ax2+bx+c=0, kde koeficienty a, b, c sa zadajú ako vstupné dáta. 

2. deterministickosť (podmienenosť) - v každom okamihu výpočtu je presne určená operácia, ktorá sa má vykonať (nič nie je ponechané na náhodu, v algoritme sa nevyskytujú náhodné operácie). 

3. rezultatívnosť (konečnosť) - výpočet podľa algoritmu sa skončí po konečnom počte krokov. 

Program je algoritmus zapísaný v programovacom jazyku. Je to postupnosť deklarácií a príkazov vyjadrujúca postup transformácie vstupných dát na výstupné.

Programovací jazyk - jazyk, ktorému počítač rozumie (t.j. jazyk, ktorému rozumie sám procesor, alebo ktorému rozumie počítač len vtedy, ak má k dispozícii príslušné programové vybavenie - prekladač). 
Programovací jazyk, ktorému rozumie priamo procesor sa nazýva strojový kód. Program v tomto jazyku tvorí postupnosť elektrických signálov kódovaných pomocou jednotiek a núl. Zápis programu je neprehľadný a rozsiahly. Vyššie programovacie jazyky sú bližšie jazyku človeka, zrozumiteľnejšie a zápis programu je stručnejší. Procesor ale tomuto jazyku nerozumie, preto je potrebný prekladač, program, ktorý preloží program z vyššieho programovacieho jazyka do strojového.


Gramatika programovacieho jazyka: tvorí ju 

1. množina symbolov prípustných pre zápis programu (napr. v Pascale sú prípustné symboly: begin, * , := , A , B , C ,...) 

2. množina pravidiel spájania týchto symbolov do jazykových konštrukcií, tzv. syntax programovacieho jazyka. (Napr. konštrukcia   A:=begin   je zložená z prípustných symbolov, ale nie je vytvorená podľa syntaktických pravidiel.) 

Sémantika programovacieho jazyka je logický význam jazykovej konštrukcie vytvorenej podľa syntaktického pravidla. 

Napr. syntaktické pravidlo    <premenná>:=<výraz>    má logický význam (sémantiku): výraz na pravej strane sa vyhodnotí a výsledok sa priradí premennej na ľavej strane.

1.2. Premenná

Počítač spracováva dáta (napr. čísla, znaky, ...), ktoré sú uložené v pamäti v tzv. premenných.

Premenná je objekt označený menom (identifikátorom), ktorý má určitú hodnotu. Hodnota premennej sa môže v priebehu výpočtu meniť.

Identifikátor (meno, názov) je reťazec (postupnosť znakov) zložený z písmen, číslic a podčiarkovníka ( _ ). Prvý znak nesmie byť číslica. Malé a veľké písmená sa nerozlišujú.

Identifikátorom sa označujú nielen premenné, ale aj ostatné objekty v programe, o ktorých sa budeme učiť neskôr: konštanty, typy, podprogramy i samotný program.


Príklady prípustných identifikátorov: A, SUMA, CISLO3, NOVY_CLEN
Príklady neprípustných identifikátorov: 3CISLO, A-BOD, C* 

Doporučenie: Názov premennej volíme tak, aby napovedal, čo daná premenná obsahuje. Ak chcem mat v premennej uložené napr. číslo poschodia, nazvem ju CISLO_POSCH, alebo C_POSCH, alebo CP.

1.3. Priraďovací príkaz

Slúži na priradenie hodnoty premennej.

Syntax: <premenná> := <výraz>


Zátvorky < > sú len pomocné znaky označujúce, že miesto symbolu v týchto zátvorkách budeme písať to, čo daný symbol vyjadruje. Napr. miesto symbolu <premenná> napíšeme konkrétnu premennú, napr. X. Všetko, čo je mimo týchto ostrých zátvoriek musí byť v príkaze napísané presne tak, ako v syntaktickom pravidle. 

<výraz> môže byť aritmetický výraz, tak ako ho poznáme z matematiky zložený z konštánt a premenných, operátorov: +(sčítanie), -(odčítanie), *(násobenie), /(delenie), okrúhlych zátvoriek,... . 

<výraz> teda môže byť: 

· konštanta (3, -25, 4.51) 

· premenná (I, CISLO5) 

· zložitejší aritmetický výraz ( I+1, A*B+5) 

Sémantika: hodnota výrazu na pravej strane sa uloží do premennej na ľavej strane.

Zatiaľ budeme do premenných priraďovať len číselné hodnoty, neskôr budeme pracovať aj s inými typmi údajov. 

Príklady priraďovacích príkazov: 

	A := 5
	do premennej A sa uloží hodnota 5

	A := B + 3
	do premennej A sa uloží hodnota o 3 väčšia než je v premennej B

	A := A + 1
	hodnota premennej A sa zvýši o 1

	A := A * 2
	hodnota premennej A sa zdvojnásobí


Príklady: 
1. V premennej A je uložená dĺžka strany štvorca. Do premennej OBV chceme priradiť číselnú hodnotu obvodu štvorca a do premennej OBS číselnú hodnotu obsahu štvorca. Použijeme príkazy:

OBV := 4 * A;

OBS := A * A;
Ak máme viac príkazov v postupnosti, oddeľujeme ich bodkočiarkou.

Do zložených zátvoriek zapíšeme začiatočný a konečný stav premenných.
2. {A, B sú rozmery obdĺžnikovej záhrady}
P := A * B;
{P je plocha záhrady} 

Úlohy pre vás - doplňte priraďovacie príkazy:

1. {R .. polomer kruhu}
?????
{OBV .. obvod kruhu, OBS .. obsah kruhu} 

2. {T1, T2, T3 .. tri teploty namerané ráno, na obed a večer}
?????
{PRIEM_TEPL .. priemerná denná teplota} 

3. {A =  }
Tento zápis znamená, že v premennej A je nejaká číselná hodnota, vo všeobecnosti sme ju označili  }
?????
?????
{A = 3 * , B = 12 * } 

4. Miško má M jabĺk a Janko J jabĺk. Zapíšte pomocou priraďovacích príkazov: 

· Janko dostal o 5 jabĺk viac ako Miško 

· Miško zjedol všetky jablká 

· Janko dal 3 jablká Miškovi 

· Janko dal Miškovi toľko jabĺk, aby mali rovnako 

5. * {A = , B = }
?????
{A = , B = } 

1.4. Príkaz vstupu

slúži na priradenie hodnoty premennej zo vstupného zariadenia (napr. z klávesnice).

Syntax: readln ( <zoznam premenných> )

<zoznam premenných> je postupnosť premenných oddelených čiarkami.

Sémantika: do premenných uvedených v príkaze uloží hodnoty zadané z klávesnice. (Ak sú to číselné hodnoty, môžu byť zadané v jednom vstupnom riadku oddelené medzerami.)

Napríklad: príkaz  readln(A,B,X3)  znamená, že počítač čaká na zadanie troch hodnôt z klávesnice. Prvú uloží do premennej A, druhú do premennej B a tretiu do premennej X3.

1.5. Príkaz výstupu

slúži na vypísanie hodnôt na obrazovku. 


Syntax: writeln ( <zoznam parametrov> )

<zoznam parametrov> je postupnosť parametrov oddelených čiarkami. Parametrom môže byť:

1. číslo (5, 4.3) 

2. premenná (A, B4) 

3. aritmetický výraz (A+4, B+C*7) 

4. text v apostrofách ('MAMA MA EMU') 

Sémantika: počítač vypíše do jedného riadku na obrazovke: 

1. dané číslo 

2. hodnotu danej premennej 

3. výslednú hodnotu daného výrazu 

4. daný text bez apostrof 

Napríklad príkaz:

writeln( 'Kruh s polomerom ' , R , ' ma plochu ' , 3.14*R*R) má 4 parametre: text, premennú, text a výraz. Za predpokladu že premenná P má hodnotu 10 vypíše na obrazovku nasledovné:

Kruh s polomerom 10 ma plochu 314

Príklad:
Čo vypíšu na obrazovku nasledovné príkazy, ak je v premennej M hodnota 5?

{ M=5 }
writeln('Misko ma M jablk.');
writeln('Misko ma ',M,'jablk.');
writeln('Misko ma M', jablk.)

1.6. Údajový typ

Údajový typ určuje: 

· množinu prípustných hodnôt 

· množinu prípustných operácií 

Každá premenná použitá v programe musí mať určený typ. Typ určuje, aké hodnoty môžeme do premennej priraďovať a aké operácie s ňou môžeme vykonávať. Pre začiatok sa zoznámime s dvoma číselnými typmi - integer a real.
1.6.1. Typ INTEGER

Množina hodnôt: celé čísla z intervalu -32 768 .. +32 767

Prípustné operácie: aritmetické ( + - * / div mod ) a relačné ( <  >  =  <>  <=  >= )

Div a mod sú operátory celočíselného delenia. Operácia div vráti celočíselný podiel a operácia mod vráti zvyšok po celočíselnom delení.

V matematike zapíšeme operáciu celočíselného delenia takto: 

27 : 6 = 4 zvyšok 3

A v Pascale takto:

27 div 6 = 4

27 mod 6 = 3

1.6.2. Typ REAL

Množina hodnôt: reálne čísla

Prípustné operácie: aritmetické ( + - * / ) a relačné ( <  >  =  <>  <=  >= )

1.7. Deklarácia premenných

Deklarovať premennú znamená určiť jej typ.

Syntax deklarácie: var <premenné> : <typ>

Napríklad: 

	var A:integer;

B,C,D:real;

E:integer
	Prípustné priradenia:
  A:=-10;

  E:=20000;

  B:=7;

  C:=4.345
	Neprípustné priradenia:
  A:=1000000;

  E:=2.5


1.8. Aritmetické operácie

Vo výraze A + B nazývame jednotlivé prvky takto: A, B sú operandy a + je operátor. Ak sú oba operandy typu integer (celé číslo), výsledok výrazu je tiež typu integer (celé číslo). Akého typu bude výsledok pre ostatné operácie a typy operandov ukazuje nasledujúca tabuľka: 

	Výraz:
	Typ operandu A:
	Typ operandu B:
	Typ výsledku:

	A + B

A – B

A * B 
	integer
integer
real
real
	integer
real
integer
real
	integer
real
real
real

	A / B
	integer,real
	integer,real
	real (vždy)

	A div B

A mod B 
	integer
	integer
	integer


Zložitejší výraz obsahujúci viac operátorov sa vyhodnocuje tak, že operácie * / div mod majú vyššiu prioritu ako + -. Prioritu ľubovoľnej operácie zvýšime pomocou okrúhlych zátvoriek.

Príklad č. 1:

Máme zadeklarované premenné: 
var A,B,C:integer;


D,E:real

Určte typy týchto výrazov: 

a) A + D

b) (A + B) * C

c) (A + B) / C

d) (A - C) * (C + E)

e) B div 5 + E

f) B div (5 + E)

g) A / B div C

Príklad č. 2: Máme zadeklarované premenné: 

var A,B,C:integer;


D,E:real

Určte, ktoré priraďovacie príkazy sú neprípustné: 

a) D := A / B

b) D := A div B

c) A := C / D

d) A := C div D

e) E := C / D

1.9. Číselné funkcie

Ak chceme číslo A umocniť na druhú, v matematike to zapíšeme: A2. V Pascale to zapíšeme pomocou funkcie SQR(A). A sa nazýva parametrom funkcie. Ak je parametrom tejto funkcie číslo 4, dostaneme ako výsledok 16. Ak je parameter typu integer, výsledok je tiež typu integer. Použitie niektorých funckií, typ ich argumentov a výsledkov popisuje táto tabuľka: 

	Popis:
	Funkcia:
	Typ parametra:
	Typ výsledku:
	Príklad:

	celá časť čísla
	TRUNC(A)
	real, integer
	integer
	TRUNC(2.7)=2

	zaokrúhlenie
	ROUND(A)
	real, integer
	integer
	ROUND(2.7)=3

	absolútna hodnota
	ABS(A)
	real, integer
	real, integer
	ABS(-7.1)=7.1

	druhá mocnina
	SQR(A)
	real, integer
	real, integer
	SQR(4)=16

	druhá odmocnina
	SQRT(A)
	real, integer
	real
	SQRT(16)=4

	sínus
	SIN(A)
	real, integer
	real
	SIN(0)=0

	cosínus
	COS(A)
	real, integer
	real
	COS(0)=1

	arcustangens
	ARCTAN(A)
	real, integer
	real
	ARCTAN(0)=0


1.10. Stavba programu

	program <meno>        
	hlavička programu

	<deklarácie>;        
	deklaračná časť

	begin             
   <príkaz>;


   <príkaz>;

   ...

   <príkaz>

end.   



	príkazová časť


Meno programu je identifikátor, ako sme si ho opísali v časti "Premenná". Názov programu volíme podľa toho, aký problém program rieši. Program na výpočet miezd môžeme napr. nazvať MZDY, alebo VYPOCET_MIEZD

Ten istý identifikátor nemôžeme použiť na označenie dvoch rôznych objektov, napr. programu i premennej. V opačnom prípade zahlási počítač chybu: „Duplicate identifier“ (duplikovaný identifikátor).

V deklaračnej časti zapisujeme deklarácie premenných (neskôr i iných objektov), ktoré budú použité v príkazovej časti. Ak by sme v príkazovej časti použili premennú, ktorú sme zabudli zadeklarovať, počítač bude hlásiť chybu: „Unknown identifier“ (neznámy identifikátor).

Príklad: Miško kráčal do školy rovnomernou rýchlosťou V metrov za sekundu. Ako dlho mu trvala cesta, ak je škola vzdialená S metrov?

Úlohou programu je pre dané vstupné hodnoty V a S vypočítať výstupnú hodnotu T - čas trvania cesty do školy. Vstupné a výstupné hodnoty je dobré zapísať si už na začiatku riešenia problému. Vstupné hodnoty program načíta, výstupné vypíše ako výsledok.

{VST: V..rýchlosť pohybu (reálne číslo), S..dráha (reálne číslo)}
{VÝS: T..čas trvania pohybu (reálne číslo)}

program CESTA_DO_SKOLY;

var V,S,T:real;

begin

  readln(V,S);

  T:=S/V;

  writeln('Miskovi trvala cesta ',T,' sekund.')

end.

Takto zapísaný program je správny. Keď si ho však spustíme na počítači, zistíme, že má určité chybičky krásy: 

1. Keď program spustíme, počítač najskôr čaká na zadanie vstupných hodnôt V a S. Nezainteresovaný užívateľ však netuší, prečo sa program hneď na začiatku "zasekol". Preto je vhodné pred každým príkazom readln vypísať na obrazovku výzvu pre užívateľa, čo má zadať, napr.: 

writeln('Zadaj rychlost a drahu!');

readln(V,S);
alebo krajšie: 

writeln('Zadaj rychlost!');

readln(V);

writeln('Zadaj drahu!');

readln(S); 

2. Keď výpočet skončí, počítač vypíše výsledok na obrazovku, nestihneme si ho ani prečítať a hneď sa obrazovka prepne do editovacieho režimu. Ak chceme, aby po vypísaní výsledku počítač počkal, musíme ho k tomu prinútiť. Ako posledný príkaz programu dáme "prázdny" readln. Tento príkaz nebude nič načítavať, len počká, až užívateľ stlačí Enter. 

3. Po spustení programu zistíme, že nám počítač vypísal výsledok asi v takomto tvare: 2.1234560000E+02 . Pre bežného používateľa je možno takýto tvar nezrozumiteľný, preto by sme ho mali preformátovať. O formátovanom výstupe sa dočítate ďalej. 

1.10.1. Formáty zápisu reálnych čísel

Reálne číslo môžeme zapísať dvoma spôsobmi: 

· bežný tvar (ako ho poznáme z matematiky), napr.: 0.00123 

· semilogaritmický tvar (ako ho poznáme z kalkulačky), napr.: 1.23E-3
Hodnota takto zapísaného čísla je: 1.23E-3 = 1,23 . 10-3 = 0,00123 

Ak používame v programe premennú typu real a necháme ju vypísať na obrazovku, počítač ju vypíše v semilogaritmickom tvare. Ak ju necháme vypísať v predpísanom formáte, vypíše ju v bežnom tvare.
Formátovaný výstup: v príkaze writeln zapíšeme za príslušný výraz dvojbodku a celkový počet miest pre výpis, resp. ďalšiu dvojbodku a počet desatinných miest.

var A:integer;

    B:real;

begin

  A:=-25;

  B:=8.4E-2;

  writeln('Cisla su', A:7 , B:10:4)

end.  

Program vypíše číslo -25 na sedem miest, t.zn. doplní štyri medzery zľava a číslo 0.084 na desať miest, z toho budú 4 desatinné, t.zn. doplní jednu nulu sprava a štyri medzery zľava: 

Cisla su    -25    0.0840

1.11. Podmienený príkaz

- príkaz, ktorého vykonanie je viazané na splnenie určitej podmienky.

Podmienený príkaz je štruktúrovaný príkaz, to znamená že je zložený z iných jednoduchších príkazov.

Syntax:

   if <logický výraz>

   then <príkaz>

   else <príkaz>

<logický výraz> je výraz, ktorého výsledná hodnota je 'pravda' alebo 'nepravda'. Môže to byť podmienka zapísaná v tvare: A < B alebo C*10 <= D  a pod.
<príkaz> je ľubovoľný príkaz (priraďovací, príkaz vstupu, výstupu, podmienený príkaz,...)

Sémantika: Ak je logický výraz (podmienka) pravdivý, vykoná príkaz na vetve 'then', ak je nepravdivý, vykoná príkaz na vetve 'else'.
Príklad: 

if A mod 2 = 0

then writeln('Cislo ',A,' je parne.')

else writeln('Cislo ',A,' je neparne.')

Tento príkaz podľa hodnoty zvyšku po delení čísla A dvoma vypíše, že číslo A je párne, resp. nepárne.

Neúplný podmienený príkaz - bez vetvy else: 

   if <logický výraz>

   then <príkaz>

Ak podmienka platí, vykoná sa uvedený príkaz, ak podmienka neplatí, príkaz sa nevykoná.

Môže sa stať, že ak podmienka platí, chceme aby sa vykonal nie jeden, ale viac príkazov. Ale na vetvu then (to isté platí pre else) môžeme dať len jeden príkaz. V tomto prípade musíme všetky príkazy, ktoré chceme mať na jednej vetve, zjednotiť do jedného zloženého príkazu - sekvencie - označenej symbolmi begin a end: 

begin

   <príkaz>;

   <príkaz>;

   ...

   <príkaz>

end

Príklad: 

{A=20, B=10}

if A<=B

then begin A:=A+1;

           B:=B-A

     end

else begin B:=B+1;

           A:=A-B

     end

{A=?, B=?}

Riešenie: podmienka A<=B sa vyhodnotí ako nepravdivá, pretože 20 nie je menšie ani rovné ako 10, preto sa vykonajú príkazy na vetve else. Konečné hodnoty premenných budú A=9, B=11.

Príklady pre vás: 

1. Janko má záhradu s rozmermi A x B metrov, Miško má záhradu s rozmermi C x D metrov. Program načíta A,B,C,D a vypíše, kto má väčšiu záhradu (s väčšou plochou). 

2. Program načíta číslo a vypíše na obrazovku jeho vlastnosti: či je celé/má desatinnú časť, kladné/záporné/nula. Napr. číslo 25.31 je kladné s desatinnou časťou, -8 je záporné celé, a pod. 

3. Miško s Jankom súťažili v hode guľou. Každý mal dva pokusy, z ktorých sa počítal ten lepší. Vytvorte program, ktorý načíta hodnoty pokusov oboch pretekárov a vypíše, ktorý z nich vyhral. 

1.12. Zložená podmienka

je zložená z jednoduchších podmienok s použitím logických operátorov AND (konjunkcia), OR (disjunkcia, alternatíva), NOT (negácia). Spôsob vyhodnocovania zložených podmienok ukazujú nasledovné tabuľky: 

	Podmienka1
	Podmienka2
	Podmienka1 AND Podmienka2

	pravda
	pravda
	pravda

	pravda
	nepravda
	nepravda

	nepravda
	pravda
	nepravda

	nepravda
	nepravda
	nepravda


	Podmienka1
	NOT Podmienka1

	pravda
	nepravda

	nepravda
	pravda


	Podmienka1
	Podmienka2
	Podmienka1 OR Podmienka2

	pravda
	pravda
	pravda

	pravda
	nepravda
	pravda

	nepravda
	pravda
	pravda

	nepravda
	nepravda
	nepravda


Pri zložených podmienkach si musíme dať pozor na prioritu operácií. Ak chceme zapísať podmienku, že číslo A je z intervalu <10;100>, musíme použiť okrúhle zátvorky:
(C>=10)AND(C<=100)

pretože priorita operácií je nasledovná: 

1. NOT 

2. AND 

3. OR 

4. < > = 

Príklady pre vás: 

1. Program načíta rok R a vypíše, či v ňom bola letná olympiáda, zimná olympiáda alebo nič. Letná olympiáda je v roku deliteľnom štyrmi, zimná je dva roky po letnej. 

2. Program načíta tri prirodzené čísla A, B, C a zistí, či môžu byť stranami trojuholníka. Ak áno, vypíše vlastnosti tohto trojuholníka - rovnoramenný, rovnostranný, pravouhlý. 

1.13. Cyklus FOR

- cyklus so známym počtom opakovaní.


Cyklus je príkaz, ktorý sa vykoná opakovane, niekoľkokrát. Ak sa dá počet opakovaní presne určiť, použijeme príkaz FOR. Ak presný počet nevieme, ale vieme, dokedy sa má opakovať, použijeme príkaz WHILE alebo REPEAT.


Syntax: 

for <premenná> := <výraz> to <výraz>

do <príkaz>

Oba výrazy i premenná musia byť typu integer. Príkaz v cykle môže byť len jeden, ak potrebujeme zopakovať viac príkazov, musíme z nich vytvoriť sekvenciu pomocou 'begin' a 'end'. Premenná sa nazýva riadiaca premenná cyklu for - zjednodušene počítadlo.

Neskôr, keď sa zoznámime aj s inými typmi, budeme ich môcť použiť pre premennú v cykle for, ale len ordinálne typy (celočíselné, char, vymenovaný typ).

Sémantika: Na začiatku sa do premennej priradí hodnota prvého výrazu (štartovacia hodnota) a po každom prechode cyklom sa jej hodnota zvýši o 1. Keď dosiahne hodnotu druhého výrazu (konečnú hodnotu), cyklus končí. Príkaz v cykle sa vykoná opakovane pre všetky celočíselné hodnoty premennej z intervalu určeného výrazmi.


Príklad: 

{ N=5 }

for i:=1 to N+5

do writeln(I)

Príkaz writeln(I) sa vykoná pre všetky hodnoty premennej I z intervalu <1;10>, t.zn. výsledok bude ten istý, ako keby sme 10-krát zapísali príkaz writeln: 

writeln(1);

writeln(2);

writeln(3);

writeln(4);

writeln(5);

writeln(6);

writeln(7);

writeln(8);

writeln(9);

writeln(10);

Ak v cykle FOR použijeme miesto symbolu to symbol downto, bude sa riadiaca premenná po každom prechode cyklom zmenšovať o 1: 

for I:=5 downto 1

do writeln(I)

Tento príkaz vypíše pod seba čísla 5,4,3,2,1.


Počet opakovaní cyklu: 

for I:=D to H

do ...

Počet opakovaní závisí od hraničných hodnôt D a H a dá sa vyjadriť výrazom: H-D+1. Ak sú hraničné hodnoty D a H rovnaké, príkaz sa zopakuje len raz. Ak je dolná hranica D väčšia ako horná hranica H, príkaz v cykle sa nevykoná ani raz. 


Príklady: 

1. Program načíta prirodzené číslo C a vypíše jeho malú násobilku v tvare:
1 x C = ...
2 x C = ...
...
10 x C = ...
2. Program načíta prirodzené číslo C a vypíše všetky jeho delitele. 

3. Program načíta prirodzené čísla A, N a vypočíta hodnotu mocniny AN. 

4. Program načíta prirodzené číslo N a vypíše všetky mocniny čísla 2 od nultej po N-tú. 

1.14. Spracovanie postupnosti

Postupnosť je skupina prvkov usporiadaných v rade za sebou. Každý prvok, okrem prvého, má svojho predchodcu a každý prvok, okrem posledného, má svojho nasledovníka. Môžeme mať napr.:

· postupnosť čísel: 5, -8, 2, 0, 20

· postupnosť mien: Jano, Jožo, Mišo, Kubo

· postupnosť farieb: modrá, zelená, červená

Počet prvkov postupnosti nazývame dĺžkou postupnosti. Každý prvok má svoje poradové číslo a postupnosť symbolicky zapíšeme takto:

a1, a2, a3, ..., aN
kde N je dĺžka postupnosti. 


My budeme zatiaľ pracovať s číselnými postupnosťami. V praxi sa často stretneme s problémom zistiť súčet alebo priemer prvkov postupnosti, zistiť počet prvkov danej vlastnosti, nájsť najväčší prvok a pod.

Na vyriešenie týchto úloh potrebujeme poznať:

- počet prvkov postupnosti (uložíme ho do jednej premennej, napr. N)

- hodnoty jednotlivých prvkov (tých je viac, potrebovali by sme na to viac premenných, ale nevieme vopred, koľko).

My však nepotrebujeme poznať naraz všetky prvky postupnosti. Môžeme si ich načítavať postupne po jednom do jednej premennej, napr. A, a priebežne ich spracovávať.

Schéma algoritmu, ktorý pracuje s N-prvkovou postupnosťou reálnych čísel: 

var N:integer;

    A:real;

begin

  writeln('Zadaj dlzku postupnosti:');

  readln(N);

  for I:=1 to N

  do begin 

       writeln('Zadaj ',I,'. cislo postupnosti:');

       readln(A);

       <spracuj cislo A>

     end;

  <vypis vysledok>

end.

Príklady: 

1. Program načíta počet detí N a N-prvkovú postupnosť nezáporných čísel a1, a2, a3, ..., aN, kde ai sú úspory I-teho dieťaťa. Zistite: 

a. celkovú hodnotu úspor detí 

b. priemernú hodnotu úspor na dieťa 

c. počet "chudobných" detí (ktoré majú nulové úspory) 

2. Miško pracoval na brigáde N dní. Prvý deň zarobil K korún. Každý ďalší deň zarobil dvakrát toľko ako v predchádzajúci deň a ešte 1 korunu navyše. 

a. Koľko zarobil v posledný deň? 

b. Koľko si zarobil celkovo? 

1.15. Hľadanie maxima (minima) postupnosti

Maximum (minimum) postupnosti je hodnota najväčšieho (najmenšieho) prvku v postupnosti.

Napr. postupnosť 5, 10, 1, 10, 2, 7 má maximum 10 a minimum 1.

Postup hľadania maxima: Postupne prejdeme postupnosťou od prvého po posledný prvok a budeme si pamätať v pomocnej premennej MAX zatiaľ nájdený najväčší prvok. T. zn. zoberieme prvý prvok a prehlásime ho za dočasné maximum. Uložíme ho do premennej MAX a skontrolujeme ostatné prvky postupnosti, či nie sú náhodnou väčšie ako naše MAX. Ak taký prvok nájdeme, musíme ho uložiť do MAX ako nové maximum.


Príklady: 

1. Daný je počet pokusov pretekára vo vrhu guľou a hodnoty jednotlivých hodov. Program vypíše hodnotu najlepšieho pokusu. 

2. Daný je počet prvkov N a jednotlivé prvky postupnosti a1, a2, a3, ..., aN. Nájdite hodnotu najväčšieho prvku a zistite, koľkokrát sa v postupnosti nachádza. 

3. Opravte predchádzajúci program tak, aby hľadal najmenší prvok. 

1.16. Cyklus WHILE

- cyklus s podmienkou na začiatku

Syntax: 

while <logický výraz>

do <príkaz>

Sémantika: kým je logický výraz (podmienka) pravdivý, opakovane vykonáva daný príkaz. Ak podmienka neplatí, cyklus končí. Ak chceme mať v cykle viac príkazov, musíme z nich vyrobiť sekvenciu pomocou 'begin' a 'end'.


Príklad: 

{A=13, B=4}

while A>=B

do A:=A-B

Aký bude stav premenných po skončení cyklu?

Začiatočný stav premenných je: A=13, B=4.
Pred vstupom do cyklu sa podmienka vyhodnotí ako pravdivá (13>=4). Vojde do cyklu a vykoná príkaz, zníži hodnotu premennej A o 4. Stav premenných po 1. prechode cyklom: A=9, B=4.
Podmienka stále platí (9>=4), preto znova zopakuje príkaz, zníži hodnotu premennej A o 4. Stav premenných po 2. prechode cyklom: A=5, B=4.
Podmienka ešte stále platí (5>=4), znova zopakuje príkaz, zníži hodnotu premennej A o 4. Stav premenných po 3. prechode cyklom: A=1, B=4.
Podmienka (1>=4) je nepravdivá, cyklus sa končí. Konečný stav premenných je: A=1, B=4. 

Na čo slúži uvedený algoritmus a aká bude konečná hodnota premennej A, ak na začiatku platí A=103, B=5?


Príklady: 

1. Dané je prirodzené číslo C. Zistite počet cifier v tomto čísle. 

2. Dané je prirodzené číslo C. Zistite jeho ciferný súčet. 

3. Miško vhodil do pokladničky 1 korunu. Druhý deň tam vhodil 2 koruny, tretí deň 3 koruny, atď. Za koľko dní si nasporí na autíčko, ktoré stojí K korún? 

4. Daná je postupnosť celých čísel predstavujúcich platy zamestnancov (ich počet nepoznáme). Ako posledný je zadaný plat riaditeľa, ktorý jediný je väčší než 50000,- Sk. Zistite koľko zamestnancov má podnik a aký je ich priemerný plat. 

5. V banke je K korún. Postupne prichádzajú požiadavky na poskytnutie úveru - t.zn. postupne sa zadávajú sumy, ktoré si chcú klienti od banky požičať. Po načítaní každej sumy program vypíše, či je žiadosť o pôžičku vybavená kladne, alebo sa jej nedá vyhovieť z dôvodu nedostatku peňazí v bankovom trezore. Program končí, ak v banke nezostane ani koruna. 

1.16.1. Zacyklenie

- opakovanie príkazu donekonečna.

Ak je podmienka cyklu WHILE stále pravdivá, cyklus nikdy neskončí. Stáva sa to vtedy, ak programátor chybne zapíše podmienku, napr.: 

{A=10}

while A>=0

do A:=A+1

Takto zapísaný cyklus nikdy neskončí, pretože premenná A bude mať stále hodnotu väčšiu ako nula. Ak sa nám v programe podarí niečo podobné, musíme spustený program násilne ukončiť pomocou kláves Ctrl+Break. Niekedy však sú s tým problémy a musíme ukončiť celý Pascal (Ctrl+Alt+Del) alebo reštartovať počítač. Pre tento prípad treba mať poslednú verziu programu uloženú na disku. PRED KAŽDÝM SPUSTENÍM PROGRAMU SI TREBA PROGRAM ULOŽIŤ NA DISK!!!

1.17. Typ CHAR

Množina hodnôt: znaky (zapisujeme ich v apostrofách, napr. 'A', '5', '@', '!', ' ' atď.)

Prípustné operácie: porovnávanie ( < > = )

Všetkých znakov je 256, usporiadané sú v tzv. ASCII tabuľke a každému je priradené poradové číslo od 0 po 255. Toto číslo sa nazýva kód znaku, resp. ordinálne číslo. Jednotlivé znaky sa porovnávajú podľa ich kódu.

Príklady kódov: 

	'A' má kód 65

'B'        66

'C'        67

...

'Z'        90
	'a' má kód 97

'b'        98

'c'        99

...

'z'        122
	'0' má kód 48

'1'        49

'2'        50

...

'9'        57


Preto platí, že: 
'A'<'a'
'A'<'B'
'9'<'A' 

Štandardné funkcie

	Popis:
	Funkcia:
	Typ parametra:
	Typ výsledku:
	Príklad:

	ordinálne číslo znaku
	ORD(C)
	znak
	celé číslo
	ORD('A')=65

	znak prislúchajúci kódu
	CHR(I)
	celé číslo
	znak
	CHR(65)='A'

	predchodca znaku
	PRED(C)
	znak
	znak
	PRED('B')='A'

	nasledovník znaku
	SUCC(C)
	znak
	znak
	SUCC('A')='B'


Príklady: 
1. Program vypíše na obrazovku veľkú abecedu: A,B,C, ... , Z 

2. Program načítava po jednom znaku a vypisuje jeho kód, pokiaľ nebude zadaná bodka. 

3. Program vypíše všetky znaky ASCII tabuľky od 0-tého po 255-ty. 

4. Program načíta veľké písmeno abecedy a vypíše to isté malé písmeno. 

Pozn.: Typ char možno použiť pre riadiacu premennú cyklu FOR. Ak chcete vypisovať na obrazovku údaje do riadku za sebou, požite miesto príkazu WRITELN len WRITE.

1.18. Cyklus REPEAT

- cyklus s podmienkou na konci

Syntax: 

repeat

  <príkazy>

until <logický výraz>

Príkazov v cykle môže byť viac a nemusíme použiť konštrukciu 'begin..end'. 

Sémantika: príkazy v cykle sa vykonávajú opakovane, pokiaľ nebude logický výraz pravdivý. Ak je nepravdivý, cyklus pokračuje.

Príklad: 

S:=0;

repeat

  readln(A);

  S:=S+A

until A=0;

writeln(S)

Čo sa vypíše na obrazovku, ak užívateľ postupne zadáva na vstup tieto čísla: 2,5,3,0,7,0 ?


Pri každom prechode cyklom sa načíta jedno číslo zo vstupu (readln(A)) a pripočíta sa k celkovému súčtu (S:=S+A). Tieto dve operácie sa opakujú dovtedy, kým číslo zadané na vstupe nebude rovné nule (A=0). Vtedy cyklus končí. Pre horeuvedenú vstupnú postupnosť nám program vypočíta 2+5+3, čiže 10.

Príklady: 

1. Daná je postupnosť kladných a záporných čísel (príjmy a výdaje) ukončená nulou. Zistite celkový príjem, celkový výdaj a konečnú bilanciu (zisk alebo strata). 

2. Daná je postupnosť známok žiaka (1..5) ukončená nulou. Zistite počty jednotlivých známok (počet jednotiek, počet dvojok,...) a priemernú známku. 

3. Dané je prirodzené číslo A. Vytvorte a vypíšte číslo B, ktoré má tie isté cifry ako A, ale v opačnom poradí. 

1.18.1. Rozdiely medzi cyklom WHILE a REPEAT

· vyhodnotenie podmienky - ak podmienka platí, cyklus while pokračuje, repeat končí. Ak podmienka neplatí cyklus while končí, ale repeat pokračuje. 

· umiestnenie podmienky - v cykle while je podmienka na začiatku, vyhodnocuje sa skôr, než sa vykoná príkaz v cykle, preto sa môže stať, že sa príkaz v cykle nevykoná ani raz, ak je podmienka hneď na začiatku nepravdivá. V cykle repeat je podmienka na konci, t. zn. najskôr sa vykonajú príkazy v cykle, až potom sa vyhodnocuje podmienka. Príkazy v cykle repeat sas teda vykonajú aspoň raz. 

· počet príkazov v cykle - v cykle while môže byť len jediný príkaz, v cykle repeat ich môže byť viac. 

Úloha: Prepíšte horeuvedený príklad (súčet postupnosti čísel ukončenej nulou) pomocou cyklu while, tak aby dával rovnaké výsledky.

1.19. Typ BOOLEAN

Množina hodnôt: logické hodnoty FALSE (nepravda) a TRUE (pravda)

Tieto hodnoty majú tiež svoje kódy, ktoré sa dajú zistiť funciou ORD. False má kód 0 a true 1.


Prípustné operácie: porovnávanie ( <>= ) a logické operácie (NOT, AND, OR) opísané v časti podmienený príkaz

Logický výraz - výraz, ktorého výsledná hodnota je pravda alebo nepravda. Napr. 

 A > B

(A<3) and (B <>0)

Premenná typu boolean a jej použitie:

Deklarácia: 

var B:boolean

Použitie v príkazoch: 

B:=true;

B:= 4<7;

B:= (4<7) and B;

if B then ...;

while not B do ...

1.19.1. Zisťovanie vlastností postupnosti

V praxi niekedy potrebujeme zistiť vlastnosť postupnosti (či je postupnosť kladná, rastúca,...). Aby sme vyriešili úlohu, potrebujeme skontrolovať každý prvok postupnosti, či spĺňa určitú podmienku. Napr. postupnosť je kladná vtedy, ak je každý jej prvok kladný. 

Úlohy tohto typu riešime tak, že predpokladáme, že postupnosť danú vlastnosť má. Potom postupne prekontrolujeme jednotlivé prvky a ak nájdeme čo len jediný, ktorý požadovanú vlastnosť nemá, náš "predpoklad" opravíme. Ak sú všetky prvky v poriadku, nič sa nemení.

Pri úlohách tohto typu je vhodné použiť signalizačnú premennú typu boolean, ktorá nám signalizuje, aká je postupnosť. Napr. premenná JEKLADNA:boolean bude obsahovať true, ak je postupnosť kladná, ak nie je kladná, bude obsahovať false.

Na začiatku predpokladáme, že postupnosť je kladná, t.zn. premennú inicializujeme nasledovne: JEKLADNA:=true . Potom v cykle postupne načítavame jednotlivé prvky postupnosti a kontrolujeme ich. Ak bude zadaný prvok nekladný, náš predpoklad sa narušil a môžeme ho zmeniť: JEKLADNA:=false. Na konci programu podľa obsahu premennej JEKLADNA vypíšeme výsledok. 


Príklady: 

1. Daný je počet prvkov N a N-prvková postupnosť celých čísel. Zistite, či je postupnosť kladná, t.zn. či sú všetky jej prvky kladné. 

2. Daný je počet prvkov N a N-prvková postupnosť celých čísel. Zistite, či je postupnosť rastúca, t.zn. či je každý jej prvok väčší než predchádzajúci (s výnimkou prvého). 

3. Dané je prirodzené číslo C. Zistite, či je to prvočíslo. 

1.20. Príkaz CASE

- viacnásobné vetvenie

Syntax: 

case <výraz> of

     <hodnoty>:<príkaz>;

     <hodnoty>:<príkaz>;

     ...

     <hodnoty>:<príkaz>;


  else <príkaz>

end
 

<výraz> je výraz ordinálneho typu (integer, char, boolean,...). 

<príkaz> je ľubovoľný príkaz, ak potrebujeme uviesť ja jednej vetve viac príkazov, použijeme konštrukciu 'begin...end'.

<hodnoty> je zoznam hodnôt takého istého typu, ako je <výraz>. Hodnoty sú oddelené čiarkami a môžu byť zapísané ako interval 'od..do'.
Vetva 'else' je nepovinná.

Sémantika: Vypočíta sa hodnota daného výrazu a porovná sa s hodnotami na jednotlivých vetvách. Ak je výraz rovný hodnote na niektorej vetve, vykoná sa príkaz na tejto vetve. Ak je rovný hodnotám na viacerých vetvách, vykoná sa ten príkaz, ktorý je uvedený skôr.


Príklad: Máme premennú A typu integer. Čo sa vypíše na obrazovku po vykonaní nasledovného príkazu? 

case A of

             3:writeln(4);

    5,10,15,20:writeln(A div 5);

        16..22:writeln(A+2);

  -2..2,4,6..9:writeln(A-1)

          else writeln(A)

end  

Ak je A=3, vykoná sa prvý príkaz a vypíše: 4

Ak je A=17, vykoná sa tretí príkaz (17 je z intervalu <16,22>) a vypíše: 19

Ak je A=20, teoreticky by vyhovovala aj druhá aj tretia vetva, ale vykoná sa len druhý príkaz a vypíše: 4

Ak je A=9, vykoná sa štvrtý príkaz a vypíše: 8

Ak je A=14, vykoná sa posledný príkaz na vetve else a vypíše: 14

Ako sa zmení výstup v horeuvedených prípadoch, ak výraz A v príkaze 'case' zmeníme na výraz A+1 ?

Príklady: 

1. Dané je prirodzené číslo z intervalu <1,5> predstavujúce známku žiaka. Program načíta číslo známky a vypíše ju slovne. 

2. Program načíta vek osoby, jej výšku v centimetroch a vypočíta ideálnu hmotnosť podľa tabuľky: 

	Vek
	Hmotnosť

	0 - 15
16 - 30
31 - 50
51 -
	vek*3.5
výška-110
výška-100
výška-90


3. Doplňte program tak, aby sa zopakoval pre N osôb (N je dané). 
4. Program načíta znak a vypíše, či je to veľké písmeno, malé písmeno, číslica, interpunkčné znamienko (.?!,) alebo niečo iné.

1.21. Vnorené cykly

Príkaz v tele cyklu môže byť ľubovoľný (priraďovací príkaz, príkaz vstupu, príkaz výstupu, príkaz vetvenia, príkaz cyklu, ...). To znamená, že príkaz, ktorý necháme vykonávať opakovane, môže byť opäť cyklus. Vznikne nám príkazová konštrukcia, ktorú nazývame vnorené cykly. 

for I:=1 to 10

do for J:=1 to 20

   do writeln('*')
Prvý cyklus sa nazýva vonkajší, vypísaný tučným je vnútorný.

Samotný príkaz... 

for J:=1 to 20

do writeln('*')

...vypíše na obrazovku 20 hviezdičiek. Ak ho necháme zopakovať 10-krát, vypíše 10-krát viac hviezdičiek, čiže 200.

Úlohy: 

1. Koľko hviezdičiek sa vypíše na obrazovku? 

for I:=1 to 5

do for J:=I to 10

   do writeln('*')

2. Koľko hviezdičiek sa vypíše na obrazovku? 

for I:=5 downto 1

do for J:=3 to I

   do writeln('*')

3. Čo sa vypíše na obrazovku? 

for I:=1 to 5

do for J:=1 to 5

   do writeln(I,J)

4. Čo sa vypíše na obrazovku? 

for I:=1 to 3

do for J:=1 to 3

   do for K:=1 to 3

      do writeln(I,J,K)

Príklady: 

1. Vypíšte na obrazovku všetky možné usporiadané dvojice znakov z množiny { A, B, C, D, E }. 

2. Vypíšte na obrazovku všetky možné kombinácie znakov z množiny { A, B, C, D, E }. 

3. Daný je počet žiakov N a počet známok na vysvedčení P. Pre každého žiaka program načíta P známok, vypočíta priemernú známku a vypíše, či bol vyznamenaný (priemer do 1.50), prospel veľmi dobre (priemer do 2.00) alebo prospel (v ostatných prípadoch). 

4. Nájdite všetky trojice prirodzených čísel A,B,C z intervalu <1,100>, ktoré sú stranami pravouhlého trojuholníka. 

5. Nájdite všetky riešenia kryptogramu (zistite, ktoré číslice reprezentujú znaky A,B,C):   BA * AB = ACA 

1.22. Výstup na textovú obrazovku

Textová obrazovka má 25 riadkov a 80 stĺpcov. Je rozdelené na 80x25 políčok, v každom políčku môže byť napísaný práve 1 znak. Aktívne políčko označuje kurzor (blikajúci podčiarkovník), ktorý sa pri výpise na obrazovku posúva ďalej. (Kurzor označuje miesto, kde začne vypisovať najbližší príkaz write/writeln. Po zápise znaku na aktívne políčko sa kurzor posunie o 1 políčko vpravo, ak je na konci riadku, posunie sa na prvé políčko v nasledujúcom riadku. Ak je kurzor v poslednom políčku posledného riadku, obsah obrazovky sa roluje o 1 riadok vyššie a kurzor sa nastaví na prvé políčko posledného riadku.

Príkazy na výpis: 

write(<parametre>) 

- vypíše údaje a nastaví kurzor za posledný vypísaný znak 

writeln(<parametre>) 

- vypíše údaje a nastaví kurzor na začiatok nasledujúceho riadku 

writeln 

- nevypíše nič, kurzor nastaví na začiatok nasledujúceho riadku 

Príklady: 

1. Dané je prirodzené číslo N. Zapíšte do riadku za sebou faktoriály: 1!, 2!, 3!, ... N! oddelené čiarkami. K- faktoriál je súčin prirodzených čísel od 1 po K: K!=1.2.3....K 

2. Dané je prirodzené číslo N. Program vypíše N riadkov s N hviezdičkami. ("Nakreslí" hviezdičkový štvorec.) 

3. Program vypíše do riadkov 1..24, toľko hviezdičiek, koľko určuje funkcia f(i)=(i-12)2/2 po zaokrúhlení (i je číslo riadku). 

Pri výstupe na obrazovku môžeme využívať príkazy z unitu CRT. V takomto prípade musíme za hlavičku programu doplniť: uses Crt; 

1.23. Unit CRT

gotoxy(x,y) 

- presunie kurzor do x-tého stĺpca a y-tého riadku. Stĺpce sú číslované 1..80, riadky 1..25. 

textcolor(F) 

- nastaví farbu písma F. F je celé číslo z intervalu <0,15>. Čísla väčšie ako 15 predstavujú blikajúce farby. 

textbackground(F) 

- nastaví farbu pozadia F. F je celé číslo z intervalu <0,7>. 

window(xl,yh,xp,yd) 

- otvorí okno so súradnicami [xl,yh] v ľavom hornom rohu a [xp,yd] v pravom dolnom rohu. Okno je aktívna časť obrazovky v ktorom platia relatívne súradnice vzhľadom na jeho ľavý horný roh [0,0]. Mimo okna sa nedajú robiť žiadne zmeny (písať, zotierať). 

clrscr 

- zotrie obrazovku, resp. okno, farbou pozadia. 

wherex, wherey 

- funkcie, ktoré vrátia prirodzené číslo - číslo stĺpca, riadku, v ktorom sa nachádza kurzor. 

delay(t) 

- procesor "čaká" t milisekúnd (keď potrebujeme spomaliť program). 

Príklady: 

1. Vyberte si zo svojich odladených programov jeden a doplňte ho o farby: na začiatku programu bude prázdna obrazovka zelenej farby. Pri načítavaní vstupných dát budú oznamy vypisované na obrazovku bielou farbou, výsledky sa budú vypisovať žltou farbou. (Pri nastavovaní farieb môžete používať aj anglické označenia: green, white, yellow). 

2. "Nakreslite" na bielej obrazovke uprostred červený obdĺžnik a rozmermi 40x15 políčok. 

3. Presúvajte pomaly po hornom okraji obrazovky nápis "AHOJ". 

4. Vytvorte na celej obrazovke farebnú reklamu na nejaký produkt. (Stačí statická, môže byť aj pohyblivá.) 

1.24. Vstup z klávesnice

Užívateľ zadáva vstupné údaje v niekoľkých riadkoch ukončených enterom . V jednom riadku môže byť zapísaných viac údajov (ak sú to čísla, oddeľujeme ich medzerami). Myslený ukazovateľ (niečo ako kurzor) označuje znak zo vstupného riadku, ktorý sa bude najbližšie čítať. Pri čítaní sa posúva ďalej.

Príkazy na čítanie: 

read(<premenné>) 

- prečíta údaje zo vstupného riadku do daných premenných a posunie ukazovateľ za posledný prečítaný znak 

readln(<premenné>) 

- prečíta údaje zo vstupného riadku do daných premenných a posunie ukazovateľ na ďalší vstupný riadok (ak ho užívateľ ešte nezadal, tak čaká) 

readln 

- posunie ukazovateľ na ďalší vstupný riadok (ak ho užívateľ ešte nezadal, tak čaká). Vstupný riadok môže byť aj prázdny (užívateľ stlačí len enter).

Príklad: 
Užívateľ zadal na klávesnici tieto vstupné riadky:
12 34
5

Aký bude obsah premenných var A,B:integer; C:char po vykonaní týchto príkazov?

read(A);

readln(C);

read(B)

{ A=    , B=    , C=     }

Funkcia EOLN : je to logická funkcia, ktorá nám signalizuje pozíciu mysleného ukazovateľa pri čítaní zo vstupného riadku. Ak je ukazovateľ na konci vstupného riadku (prečítal sa posledný znak), funkcia vráti TRUE, inak vráti FALSE.

Túto funkciu môžeme využiť pri čítaní postupnosti údajov, ktorá je daná v jednom vstupnom riadku, pričom počet prvkov vopred nepoznáme. Použijeme túto schému programu: 

while not eoln

do begin read(prvok);

         <spracuj prvok>;

   end

Príklad: Daná je postupnosť znakov v jednom vstupnom riadku. Pri zadávaní sú na obrazovke štandardné farby - čierne pozadie, biele písmo. Vypíšte tú istú postupnosť do druhého riadku žltou farbou. 

uses crt;

var znak:char;

begin

  textbackground(black);

  textcolor(white);

  writeln('Zadaj postupnost znakov:');

  while not eoln

  do begin read(znak);

           textcolor(yellow);

           write(znak)

     end

end.

Príklady: 

1. Daná je postupnosť znakov v jednom vstupnom riadku. Program vypíše, koľko je v postupnosti

a) písmen b) číslic c) aritmetických znamienok 

2. Daná je postupnosť znakov v jednom vstupnom riadku. Vypíšte tú istú postupnosť do druhého riadku tak, že:

a) každé 'y' zmeníte na 'i  b) za každé 'x' vložíte 'y' 

3. Dané je prirodzené číslo A. Bez použitia číselnej premennej

a) zistite, či je deliteľné 5 b) zistite, či je deliteľné 100

c) vypíšte jeho desaťnásobok. 

4. Daná je postupnosť znakov. Zistite, či je to identifikátor. 

1.25. Knižnica CRT - vstup z klávesnice

readkey 

- znaková funkcia, ktorá čaká na stlačenie klávesy a vráti znak tejto klávesy. Po stlačení klávesy sa znak nevypisuje na obrazovku. 

keypressed 

- logická funkcia, ktorá vráti TRUE, ak bola stlačená klávesa, inak vráti FALSE 

Príklady: 

1. Do stredu obrazovky vypíšte text AHOJ a opakovane ho prekresľujte farbami od 1 do 15 pokiaľ užívateľ nestlačí nejakú klávesu. 

2. Do stredu obrazovky vypíšte text AHOJ. Program opakovane načítava znak stlačenej klávesy. Ak užívateľ stlačí 'C', nápis sa prepíše červenou farbou, ak stlačí 'M' - modrou, 'Z' - zelenou, ... 'K' - program končí. 

Klávesy môžeme rozdeliť na: 

· znakové ( A, @ , + , 3 , medzera , ...) - ich stlačením užívateľ zadáva zobraziteľný znak 

· funkčné (Enter, Esc, šípky, F1, Insert, ...) - ich stlačením sa vyvoláva nejaká funkcia (napr. Esc má väčšinou funkciu storno, F1 - help, ...) 

· prepínacie (Alt, Shift, Ctrl) - používajú sa v kombinácii s inou klávesou a menia jej funkciu, resp. znak 

Iné rozdelenie kláves: 

· jednobajtové - ich stlačením sa zadáva jeden znak ( A, 3, + , Enter, Esc, ...) 

· dvojbajtové - ich stlačením sa zadávajú dva znaky, z ktorých prvý je znak s kódom 0 (šípky, F1, ...) 

Znaky môžeme zapisovať (okrem bežného spôsobu s apostrofami, napr. 'A') aj pomocou ich kódov: #<kód> (napr. #65). Tento spôsob sa používa hlavne pre nezobraziteľné znaky (napr. klávesa Esc zadáva znak #27).

Prehľad znakov niektorých kláves: 

	Esc
	jednobajtová
	#27

	Enter
	jednobajtová
	#13

	šípka hore
	dvojbajtová
	#0 #72

	šípka vľavo
	dvojbajtová
	#0 #75

	šípka vpravo
	 
	 

	šípka dole
	 
	 

	Home
	 
	 

	End
	 
	 

	PgUp
	 
	 

	PgDn
	 
	 

	Insert
	 
	 

	Delete
	 
	 

	F1
	 
	 

	Ctrl F1
	 
	 


Príklad: Program opakovane načítava stlačenú klávesu a vypisuje, či je jednobajtová alebo dvojbajtová a vypíše jej kód, resp. kódy. S pomocou tohto programu doplňte horeuvedenú tabuľku. Program končí po stlačení Esc. Doplňte chýbajúce časti označené "....." 
uses Crt;

var ZN1, ZN2:char;

begin

  repeat

    ZN1:=readkey;

    if ZN1=.....

    then begin ZN2:=readkey;

               writeln('Klavesa je dvojbajtova, jej kody su: 0 a ',.....)

         end

    else writeln('Klavesa je jednobajtova, jej kod je: ',.....)

  until ZN1=.....

end. 

Príklady: 

1. Vypíšte do stredu obrazovky text AHOJ. Program opakovane načítava stlačené klávesy a podľa šípok posúva text o riadok vyššie/nižšie, o stĺpec doľava/doprava. Po stlačení Esc program končí. 

1.26. Kontrolovaný vstup

Vstupná podmienka algoritmu určuje, aké vlastnosti majú mať údaje, ktoré dostane algoritmus na vstupe. Napr: "Dané je prirodzené číslo N väčšie ako 1.", "Daná je postupnosť kladných reálnych čísel ukončená nulou." Pre užívateľa to znamená, že ak chce, aby mu algoritmus dal na záver správne výsledky, musí mu na vstupe zadať údaje vyhovujúce vstupnej podmienke. Zadanie nesprávneho údaju na vstupe môže spôsobiť, že program skončí predčasne s chybovým hlásením (program "spadne") alebo nám dá na výstupe zlé výsledky. Dobrý program by mal byť zabezpečený proti chybám užívateľa a kontrolovať každý údaj zadaný na vstupe. V prípade zadania nesprávneho údaju sa musí vstup zopakovať.

Chyba vstupného údaju môže byť:

- v type (napr. užívateľ mal zadať celé číslo a zadal znak)

- v hodnote (napr. užívateľ mal zadať číslo z intervalu <1,10> a zadal 20)


Schéma kontrolovaného vstupu: 
repeat

    writeln('Zadaj...');

    readln(U);

    if <údaj U je zlý>)

    then writeln('Zle - zadaj znova')

until <údaj U je je dobrý>

Zatiaľ budeme kontrolovať len správnosť hodnoty.

Príklad: 
Naprogramujte kontrolovaný vstup celého čísla z intervalu <-10,10>.

1.27. Podprogram

[image: image2.wmf] 

Podprogram je logicky ucelená časť programu - postupnosť príkazov, ktorá rieši nejaký podproblém a je označená vlastným menom (identifikátorom).

Napr.: Našou úlohou je vytvoriť program, ktorý nakreslí na obrazovku obdĺžnikovú dúhu, v hornej časti v strede vypíše "PRŠÍ PRŠÍ" a v strednej časti nechá rolovať riadky, v ktorých sú striedavo vypísané bodky a medzery podľa vzoru: 
Túto úlohu môžeme rozdeliť na čiastkové úlohy (podproblémy): nakresliť dúhu, urobiť nápis a urobiť dážď (rolujúce riadky s bodkami a medzerami). Každý podproblém bude riešiť nejaký podprogram, nazvime ich: DUHA, NAPIS, DAZD. Ak by tieto "príkazy" počítač poznal, náš program by bol veľmi jednoduchý: 

begin

  DUHA;

  NAPIS;

  DAZD;

end.

Príkazy, ktoré sme v programe použili, nie sú súčasťou jazyka Pascal, preto ich musíme počítaču presne "vysvetliť", musíme ich zadeklarovať.

Deklarácia podprogramu (procedúry): 

procedure <identifikátor>;

<deklaračná časť>;

begin

  <príkazy>;

end

Podprogram má rovnakú stavbu ako program, rozdiel je len v hlavičke (miesto 'program' píšeme 'procedure'). Za symbolom 'end' nedávame bodku, ale bodkočiarku, ktorá slúži ako oddeľovač od ďalšej časti.


Umiestnenie deklarácie podprogramu: v deklaračnej časti hlavného programu (tam, kde je umiestnená deklarácia premenných). Zaužívané poradie je, že najskôr deklarujeme premenné a po nich podprogramy, ale môže to byť aj naopak.

Volanie podprogramu: v hlavnom programe, tam, kde chceme, aby sa vykonali príkazy podprogramu, napíšeme len jeho meno. Takýto zápis sa berie ako jeden príkaz.


Náš program bude nakoniec vyzerať takto: 

program APRIL;

    procedure DUHA;

    begin

      ...

    end;

    procedure NAPIS;

    begin

      ...

    end;

    procedure DAZD;

   begin

    window(20,7,60,19); {súradnice okna treba prispôsobiť}

      textbackground(lightgray);

      textcolor(black);

      clrscr;

      repeat

        write('. ');

      until keypressed

   end;

Begin
   DUHA;

   NAPIS;

   DAZD

end.

Úloha pre vás: Doplňte chýbajúce príkazové časti podprogramov.

Výhody použitia podprogramov: 

· prehľadnejší program 

· možnosť ladiť program postupne po častiach 

· stručnejší zápis programu (ak sa nejaká postupnosť príkazov opakuje na viacerých miestach programu, stačí ju zapísať len raz - zadeklarovať podprogram - a volať potrebný počet krát) 

1.27.1. Lokálne a globálne premenné

Aj podprogram môže mať svoje vlastné premenné deklarované v jeho deklaračnej časti. Tieto premenné sa nazývajú lokálne a môže ich používať len tento podprogram. Okrem nich môže používať aj premenné z hlavného programu - globálne. Ak existuje lokálna aj globálna premenná s tým istým menom, sú to dve rôzne premenné a v podprograme je použiteľná len lokálna (globálna je zatienená).

Napríklad: 

program HLAVNY;

var A,B:integer;

    procedure PROC;

    var B,C:integer;

    begin

       A:=1; B:=2; C:=3

    end;

begin

   A:=4; B:=5;

   PROC;

   writeln(A,B)

end.

Aké hodnoty vypíše tento program? Čo vypíše, ak posledný príkaz zmeníme na writeln(A,B,C) ?


Príklady: 

1. Dané je prirodzené číslo N. "Vykreslite" na obrazovku štvorec so stranou N, po obvode ktorého budú hviezdičky a vnútri medzery. Vytvorte podprogram, ktorý vypíše do riadku N hviezdičiek a druhý podprogram, ktorý vypíše do riadku hviezdičku, za ňou (N-2) medzier a na koniec hviezdičku. S pomocou týchto podprogramov vytvorte hlavný program. 

1.27.2. Používanie premenných a podprogramov

Pre každý objekt v programe (premenné i podprogramy) platí, že musí byť deklarovaný skôr, ako je použitý. T. zn. ak chceme premennú z hlavného programu použiť v nejakom podprograme, musí byť deklarovaná skôr než tento podprogram. Ak chceme, aby jeden podprogram mohol volať iný podprogram, musí byť volaný deklarovaný skôr ako volajúci.

1.28. Ponukový program

Je to program, ktorý vypíše ponuku niekoľkých služieb, užívateľ si niektorú z nich vyberie a program ju zrealizuje. Toto všetko opakuje, kým užívateľ nezadá pokyn na ukončenie.

Schéma ponukového programu: 

repeat

    writeln('1. sluzba');

    writeln('2. sluzba');

    ...

    writeln('N. sluzba');

    writeln('0. koniec');

    writeln('Zadaj cislo sluzby, ktoru si si vybral:');

    readln(VYBER);

    case VYBER of

       1:<volaj procedúru 1>;

       2:<volaj procedúru 2>;

       ...

       N:<volaj procedúru N>

    end;

until VYBER=0

Kvôli prehľadnosti programu je dobré každú službu realizovať zvláštnym podprogramom.


Postup ladenia ponukového programu: 

Ponukový program je rozsiahlejší a je dobré ladiť ho po častiach - po jednotlivých procedúrach. T.zn. najskôr vyrobíme hlavný program s prázdnou ponukou služieb. Doplníme doň prvú službu a procedúru, ktorá ju realizuje. Program spustíme, odskúšame, či pracuje tak, ako treba, príp. opravíme chyby. Doplníme ďalšiu službu a jej procedúru atď.
Príklady: 

1. Vytvorte ponukový program, ktorý opakovane načítava celé číslo. Ak užívateľ zadá 1, "vykreslí" slovenskú zástavu, 2 - francúzsku, 3 - maďarskú, atď., ak zadá 0 - skončí. (Zástavy kreslite ako farebné pruhy - obdĺžniky.) 

2. Ponukový program načíta prirodzené číslo N a ponúkne užívateľovi nakreslenie týchto geometrických útvarov z hviezdičiek: 1. štvorec so stranou N, 2. pravouhlý trojuholník s odvesnami dĺžky N, 3. rovnoramenný trojuholník s výškou N, 4. kosoštvorec so stranou N. 

*****  *          *    *****

*****  **        ***    *****

*****  ***      *****    *****

*****  ****    ********   *****

*****  *****  **********   *****

1.29. Typ pole

Pole je dátová štruktúra - obsahuje viacero dát nejakým spôsobom usporiadaných. Pole pozostáva z niekoľkých prvkov rovnakého typu usporiadaných v rade za sebou. Ich počet je pevne určený v deklarácii. Z uvedeného vyplýva, že pole ako dátová štruktúra má tieto vlastnosti:

- lineárna (prvky sú usporiadané v rade za sebou)
- homogénna (všetky prvky sú rovnakého typu)
- statická (počet prvkov je stály, nemenný)
Pole si môžeme graficky znázorniť takto: 

	1234
	1240
	1265
	1220
	1270
	1301
	1400
	1350
	1360
	1386
	1420
	1421


V takejto dátovej štruktúre môžeme mať uloženú informáciu o stave účtu v banke počas 12 mesiacov roka. (V prvom prvku je stav účtu v januári, v druhom prvku stav účtu vo februári atď.) Toto by bolo pole čísel.

Okrem poľa čísel si môžeme vytvoriť napr. pole mien detí: 

	Michal
	Zuzka
	Andrej
	Vilma


Alebo pole znakov, v ktorom môžu byť zapísané správne odpovede na testové otázky (A, B, C, D): 

	A
	C
	B
	A
	D
	A


Prvý prvok poľa obsahuje správnu odpoveď na prvú testovú otázku, druhý prvok - na druhú... atď.

Celé pole je jedna premenná označená jedným identifikátorom. Jednotlivé prvky poľa sú rozlíšené indexami. Index si môžeme predstaviť ako poradové číslo. 

	1
	2
	3
	4
	5
	6

	A
	C
	B
	A
	D
	A


Deklarácia premennej typu pole: 
var <premenná>:array [<typ indexu>] of <typ prvkov>

Typ indexu je ordinálny (celočíselný, char, boolean,...) a zvyčajne sa zapisuje ako interval od..do. Typ prvkov je ľubovoľný.


Nazvime polia, ktoré sme si graficky znázornili vyššie, UCET, DETI, TEST. Ich deklarácia bude vyzerať nasledovne: 

var UCET:array[1..12]of real;

    DETI:array[1..4]of string;

    TEST:array[1..6]of char

Typ string sa naučíme neskôr. V premennej typu string môže byť uložené celé meno (postupnosť znakov). 


Indexy nemusia byť vždy len celé čísla a nemusia vždy začínať od čísla 1. Napr.: 

var A:array[-5..5] of integer;

    B:array['a'..'z'] of boolean

Sprístupnenie prvku poľa: <premenná>[<index>]

Napr.: UCET[1] je prvý prvok poľa UCET, DETI[4] je posledný prvok poľa DETI.

Ak by sme ako index použili hodnotu mimo intervalu určeného v deklarácii, napr. UCET[0] alebo DETI[5], počítač by vyhlásil chybu: „Range check error“.

Ak vám aj napriek chybe v programe počítač nevypísal chybové hlásenie, treba si hlavnom menu Pascalu, v položke Options - Compiler zapnúť voľbu "Runtime errors: [X] Range checking."


Práca s prvkami poľa: s jednotlivými prvkami pracujeme ako s premennými príslušného typu. (S prvkom UCET[1] pracujeme ako s premennou typu real, s prvkom TEST[2] ako s premennou typu char.) 

UCET[1]:=123.50;

UCET[2]:=UCET[1]*2;

I:=3;

writeln(DETI[I]);

if TEST[1]='A' then P:=P+1

POZOR: premennú typu pole nemôžeme načítavať ani vypisovať: readln(UCET); writeln(UCET) 

Pole načítavame aj vypisujeme po jednotlivých prvkoch: 

	readln(UCET[1]);               

readln(UCET[2]);

readln(UCET[3]);

...

readln(UCET[12])
	alebo jednoduchšie: 

for I:=1 to 12

do readln(UCET[I])


Príklad: 
1. Daná je postupnosť desiatich celých čísel. Vypíšte tú istú postupnosť:
a) odzadu (prvky v opačnom poradí)
b) bez nulových prvkov
c) najskôr záporné a potom nezáporné prvky 

V praxi väčšinou dopredu nevieme, aké veľké pole budeme potrebovať. Úloha môže byť zadaná takto: Dané je prirodzené číslo N a N-prvková postupnosť reálnych čísel... . Hodnota N sa načíta až po zahájení programu, ale pole musí byť deklarované skôr. Ak by sme pole deklarovali takto: var A:array[1..N]of real; počítač by nám pri premennej N vyhlásil chybu: „Cannot evaluate this expression“. (Nemôžem vyhodnotiť tento výraz - nepoznám jeho hodnotu.) Rozsah indexov musí byť v deklarácii určený konštantou.

Riešime to tak, že odhadneme maximálnu hodnotu N a dĺžku poľa určíme podľa tohto odhadu. Napr.: Daný je počet žiakov v triede N a N-prvková postupnosť reálnych čísel, predstavujúca ich priemerné známky... Odhadneme, že v triede nebude viac ako 40 žiakov a pole zadeklarujeme nasledovne:

 var A:array[1..40]of real
Alebo je horná hranica vstupných hodnôt určená priamo v zadaní: Dané je prirodzené číslo N (N<=100) a N-prvková postupnosť reálnych čísel.... Deklarácia teda bude:

 var A:array[1..100]of real.
Ak by sme to s dĺžkou poľa prehnali, napr.:

 var A:array[1..20000]of real; 

počítač vyhlási chybu: „Structure too large“. (Príliš veľká dátová štruktúra - nezmestí sa do vyhradenej časti pamäte.)


Príklady: 

1. Daný je počet zamestnancov N (N<=100) a N-prvková postupnosť prirodzených čísel, predstavujúcich platy zamestnancov. Zistite ich priemerný plat a počet zamestnancov, ktorí majú podpriemerný plat. 

2. Daný je počet dní v mesiaci N a N-prvková postupnosť celých čísel predstavujúca teploty namerané v jednotlivých dňoch o 12,00 hod. Nájdite hodnotu najnižšej teploty v danom období a vypíšte čísla dní, v ktorých bola nameraná (môže ich byť viac). 

3. Daná je veta - postupnosť znakov - v jednom riadku. Zistite, či je to palindróm (po vynechaní medzier a interpunkčných znamienok, s ignorovaním rozdielu medzi malými a veľkými písmenami, je symetrická, t.zn. odpredu i odzadu znie rovnako). Napr. "Kobyla ma maly bok." je palindróm. 

4. Daný je počet detí N (N<=50) a postupnosť N reálnych čísel predstavujúcich úspory jednotlivých detí. Program načíta vstupné dáta a realizuje nasledovnú ponuku služieb:
a) výpis stavu úspor v tvare: 

   1. dieťa:   12,50 Sk

   2. dieťa:    8,10 Sk

   ...

   N. dieťa:  222,00 Sk

   --------------------

   Spolu:     xxx,xx Sk

b) zvýšiť každému dieťaťu hodnotu úspor o K korún (K je dané)

c) uskutočniť zbierku po K korún (K je dané): každé dieťa dá do zbierky K korún, ak má menej, tak dá všetko, čo má. Vypíše sa hodnota vyzbieranej sumy.

d) zúročiť úspory - pripočítať každému dieťaťu k jeho sume úroky, ktorých percento je určené tabuľkou:

suma do 10,- Sk: 1%

suma do 50,- Sk: 2%

suma nad 50,-Sk: 3%

e) zaevidovať nové dieťa (načíta sa jeho suma, ktorú vkladá do banky a zaradí sa na koniec poľa)

f) vylúčiť I-te dieťa z evidencie (I je dané)

g) zaevidovať protekčné dieťa (načíta sa jeho suma, ktorú vkladá do banky a zaradí sa na začiatok poľa) 

1.30. Deklarácia konštánt

Konštanta je konkrétna hodnota určitého typu, ktorá sa môže vyskytovať v programe na rôznych miestach. Napr. celočíselná konštanta 20 sa môže vyskytovať v deklarácii poľa, v priraďovacom príkaze, v podmienke a pod.: 

var A:array[1..20]of integer;

begin

  ...

  I:=I+20;

  ...

  if N>20 then writeln('POZOR!');

  ...

end.

Podobne znak '.' ako konštanta typu char: 

begin

  ...

  repeat

    read(znak);

    if znak<>'.' then ...;

  until znak='.';

  ...

end.

Ak sa rozhodnem napr. v prvej ukážke zväčšiť rozsah poľa na 100, musím opraviť v programe všetky konštanty s hodnotou 20, súvisiace s rozsahom poľa, na 100. V rozsiahlejšom programe to môže byť problém. Preto je výhodnejšie túto konštantu pomenovať, napr. ROZSAH a v programe používať len tento indentifikátor. Samozrejme, skôr než ho použijeme, musíme ho zadeklarovať: const ROZSAH=20. Deklaráciu umiestnime do deklaračnej časti programu, zvyčajne pred deklaráciu premenných, ale nie je to podmienkou. Podobne si môžeme zadeklarovať konštantu typu char: const KONCOVY_ZNAK='.' . Naše ukážky programov budú potom vyzerať takto: 

const ROZSAH=20;

var A:array[1..ROZSAH]of integer;

begin

  ...

  I:=I+ROZSAH;

  ...

  if N>ROZSAH then writeln('POZOR!');

  ...

end.

Podobne znak '.' ako konštanta typu char: 

const KONCOVY_ZNAK='.'

begin

  ...

  repeat

    read(znak);

    if znak<>KONCOVY_ZNAK then ...;

  until znak=KONCOVY_ZNAK;

  ...

end.

Ak sa teraz rozhodneme zväčšiť rozsah poľa na 100, zmenu stačí urobiť len na jednom mieste programu, v deklarácii konštanty: const ROZSAH=100 .

Syntax deklarácie konštanty: 

const <identifikátor>=<konštanta>

Takto deklarujeme konštanty jednoduchých typov (integer, real, char, boolean, ...).

Niektoré konštanty sú súčasťou Pascalu, nemusíme ich deklarovať, môžeme ich len používať. Jednou z nich je konštanta PI (Ludolfovo číslo 3,14...). 

1.31. Deklarácia typu

Ak používame v programe štrukturovaný typ (zatiaľ poznáme len typ pole), zvykneme ho pomenovať a v programe pri deklarácii premenných používať len jeho identifikátor.

Syntax deklarácie typu: 

type <identifikátor>=<definícia typu>

Napríklad: 

	type TPOLE=array[1..100]of real

Takýto typ môžeme použiť pri deklarácii premennej typu pole:      

var A:TPOLE
	Je to to isté ako keby sme priamo deklarovali: 

var A:array[1..100]of real


Výhodou deklarovania typu je, že ho zadeklarujeme len raz a môžme ho použiť na rôznych miestach programu pri deklarácii premenných - globálnych i lokálnych. Keď sa neskôr naučíme používať podprogramy s parametrami, deklarácia typov bude dokonca nevyhnutná.

Doporučenie:

Tak ako názvy premenných volíme podľa toho, čo obsahujú, aj názvy typov by mali korešpondovať s premennými, pri deklarácii ktorých budú použité. Zvyčajne sa názov typu začína písmenom T. Napríklad, v programe zadeklarujeme premennú typu pole, v ktorej budú uložené počty bodov, ktoré získal žiak v písomke za jednotlive otázky: 

const POCET_OTAZOK=20;

type TBODY=array[1..POCET_OTAZOK]of integer;

var BODY:TBODY

Úloha na premýšľanie:

Ak máme v programe dve premenné rovnakého typu (napr. A,B), môžeme priradiť obsah jednej premennej do druhej (napr. A:=B). Typ premenných môže byť ľubovoľný (aj pole). Napr.: 

var A,B:array[1..100] of real;

begin  A:=B  end.

Ak použijeme iný spôsob deklarácie premenných, môže nám počítač pri príkaze A:=B hlásiť chybu "Type mismatch" (Nezhoda typov). V ktorej z týchto ukážok k tomu dôjde a prečo?

a) var A:array[1..100] of real;

       B:array[1..100] of real;

   begin  A:=B  end.

b) type pole=array[1..100] of real;

   var A:pole;

       B:pole;

   begin  A:=B  end.

c) type pole=array[1..100] of real;

   var A,B:pole;

   begin  A:=B  end.

d) type pole=array[1..100] of real;

   var A:pole;

       B:array[1..100] of real;

   begin  A:=B  end.

1.32. Vymenovaný typ

Niekedy potrebujeme pracovať v programe s inými hodnotami než sú čísla, znaky, logické hodnoty a pod. Skrátka, žiaden zo štandardných typov sa nám nehodí. Môžeme si vytvoriť vlastný typ s vlastnou množinou hodnôt. Každú hodnotu označíme nejakým identifikátorom a vymenujeme ich v deklarácii typu.


Syntax deklarácie vymenovaného typu: 

type <identifikátor typu>=(<zoznam identifikátorov hodnôt oddelených čiarkami>)

Napríklad potrebujeme v programe pracovať s hodnotami predstavujúcimi dni v týždni. Označíme ich PONDELOK, UTOROK, STREDA, STVRTOK, PIATOK, SOBOTA a NEDELA. Typ s touto množinou hodnôt nazveme TDEN a zadeklarujeme ho: 

type TDEN=(PONDELOK, UTOROK, STREDA, STVRTOK, PIATOK, SOBOTA, NEDELA);

var DEN:TDEN

Zároveň sme zadeklarovali premennú DEN, do ktorej môžeme priraďovať hodnoty zo sedemprvkovej množiny hodnôt typu TDEN, napr. DEN:=STREDA.

Každá hodnota v množine má pridelený kód (celé číslo) počnúc od nuly. Napr. kód hodnoty STREDA je 2. Pre hodnoty tohto typu môžeme používať funkcie ORD, SUCC, PRED, tak ako sme ich používali pre typ char. Hodnoty vymenovaného typu nemôžeme načítať ani vypísať: readln(DEN); writeln(DEN) Používame ich kvôli lepšej čitateľnosti programu. 

1.33. Typ interval

Množina hodnôt tohto typu je podmnožinou iného ordinálneho typu (char, boolean, celočíselný a vymenovaný typ) a je určená dolnou a hornou hranicou.


Syntax deklarácie typu interval: 

type <identifikátor typu>=<dolná hranica>..<horná hranica>

Príklady deklarácií intervalov a ich použitie pri deklarácii premenných: 

type TDEN=(PONDELOK, UTOROK, STREDA, STVRTOK, PIATOK, SOBOTA, NEDELA);

     TZNAMKY=1..5;

     TABECEDA='a'..'z';

     TPRAC_DNI=PONDELOK..PIATOK;

var Z:TZNAMKY;

    ZNAK:TABECEDA;

    DEN:TPRAC_DNI

Typ TZNAMKY je podmnožinou typu INTEGER. S premennou Z pracujeme ako s premennou typu INTEGER, ale priraďovať do nej môžeme len hodnoty od 1 do 5. Podobne TABECEDA je podmnožinou typu CHAR a TPRAC_DNI je podmnožinou typu TDEN.

Interval sa používa hlavne pri deklarácii poľa: 

const MAX=100;

type TINDEX=1..MAX;

     TPOLE=array[TINDEX] of integer;

var A:TPOLE

Príklad: 

1. Naprogramujte obeh hviezdičky okolo obrazovky v smere hodinových ručičiek, pokiaľ užívateľ nestlačí ľubovoľnú klávesu. (Pracujte s riadkami 1..25 a stĺpcami 1..79, posledný stĺpec ignorujte.) 

Riešenie:
V cykle budeme opakovane vypisovať a zotierať hviezdičku, pričom jej súradnice sa budú po každom kroku meniť. Aktuálnu pozíciu hviezdičky si budeme pamätať v premenných X, Y, predstavujúcich súradnice hviezdičky. Hrubá schéma programu bude vyzerať nasledovne: 

repeat

   <vypíš hviezdičku>;

   <počkaj>;

   <zotri hviezdičku>;

   <zmeň súradnice>

until keypressed

Ako sa budú meniť súradnice? Raz sa bude meniť x-ová, raz y-ová súradnica - raz sa bude zväčšovať o 1 a raz zmenšovať, podľa toho, aký má hviezdička práve smer (východ, juh, západ, sever). Aby sme vedeli, ktorú súradnicu treba zmeniť a ako, budeme si ešte pamätať aktuálny smer v premennej typu TSMER: 

type TSMER = (VYCHOD,JUH,ZAPAD,SEVER);

var S:TSMER;

    X,Y:integer;

Program bude vyzerať nasledovne - doplňte chýbajúce časti: 

begin

  {inicializácia premenných - doplňte ďalšie}

  X:=1; Y:=1;

  repeat

    {vypíšeme hviezdičku na pozíciu [X,Y] - doplňte príkazy}

    delay(50);

    {zotrieme hviezdičku}

    gotoxy(X,Y);

    write(' ');

    {zmeníme súradnice}

    if S=VYCHOD then if X<79 then X:=X+1 else S:=JUH;

    if S=JUH    then  {doplňte}

  until keypressed

end.

1.34. Typ reťazec

Reťazec je postupnosť znakov, ktorú zapisujeme medzi apostrofy. Počet znakov určuje dĺžku reťazca.

Napr.: 'FERO' je reťazec dĺžky 4

'PD 123AB' je reťazec dĺžky 8 (počíta sa aj medzera)

'@' je reťazec dĺžky 1

'' je reťazec dĺžky 0 - prázdny reťazec

Deklarácia premennej typu reťazec: 

var <premenná>:string[<maximálna dĺžka reťazca>]

Maximálna dĺžka reťazca určuje, aký najdlhší reťazec môže byť v premennej uložený. Je to celé číslo z intervalu <0,255>. Táto časť deklarácie je nepovinná, ak ju vynecháme, počítač ju automaticky doplní na 255: 

var <premenná>:string
Napr.: 

var R:string[10];

    S:string[2];

    T:string;

Operácie s reťazcami: 

a. priradenie 

R:='ABCD';

S:=R  {ak sa priraďovaný reťazec nezmestí do premennej, skracuje sa sprava}

b. porovnanie pomocou operátorov <  >  =  <>  <=  >=
Dva reťazce sa porovnávajú podľa prvých znakov, napr.: 'ABDF'<'BBC', pretože 'A'<'B'.

c. Ak sú prvé znaky zhodné, porovnáva sa podľa najbližšej dvojice znakov, ktoré sú rôzne, napr.: 'ABCDE'<'ABCE', pretože 'D'<'E'.

d. Ak je jeden z reťazcov kratší a chýba mu ďalší znak na porovnanie, považuje sa žiadny znak za menší než ľubovoľný iný znak, napr.: 'ABC'<'ABCD', pretože ľavý reťazec nemá štvrtý znak. 

e. zreťazenie pomocou operátora +. Zreťazenie je spojenie dvoch reťazcov do jedného. Napr.: 'SKODA ' + 'FABIA' = 'SKODA FABIA'

f. načítanie - reťazce môžeme načítavať pomocou príkazov read, readln, ale zatiaľ budeme používať len readln. 

g. výpis - reťazce môžeme vypisovať na obrazovku pomocou príkazov write, writeln. 

Príklady: 

var R,S,T:string[6];

begin R:='MYŠ';

      S:='LIENKA';

      if S<R

      then T:=R+S

      else T:=S+R;

      writeln(T)

end.

Čo vypíše tento program na obrazovku? 

1. Dané je prirodzené číslo N a N-prvková postupnosť mien. Načítajte vstupné údaje a vypíšte na obrazovku mená, ktoré sa začínajú na 'A'. 

2. Daný je reťazec Z predstavujúci slovný základ, prirodzené číslo N a N-prvková postupnosť reťazcov, predstavujúcich predpony. Vypíšte všetky možné slová, ktoré vzniknú z predpôn a základu.
Doplnok: Napočítajte aj prirodzené číslo M a M-prvkovú postupnosť reťazcov - prípon. Vypíšte všetky možné slová, ktoré vzniknú zo základu, predpôn a prípon.Napr.: 3 predpony: 'O','ZÁ','VÝ', 2 prípony: 'A','NÍK', základ: 'HRAD' vzniknuté slová: 'OHRADA', 'ZÁHRADA', 'VÝHRADA', 'OHRADNÍK', 'ZÁHRADNÍK', 'VÝHRADNÍK' 

3. Dané je prirodzené číslo N. Uložte do premennej S:string reťazec zložený z N výkričníkov. 

4. Daný je reťazec predstavujúci číslo v semilogaritmickom tvare, ktorého formát je <celé číslo>E<prir.číslo> (napr.: -123E5). Program vypíše toto číslo v bežnom tvare (napr.: -12300000). 

1.35. Vyhľadávanie

Pri práci s dátovými štruktúrami sa často stretneme s potrebou vyhľadať jeden alebo viac prvkov danej vlastnosti. Napr. z postupnosti žiakov chceme vypísať všetkých vyznamenaných, v postupnosti evidenčných záznamov chceme nájsť záznam o osobe s menom 'Cyril', v rodokmeni kráľovského rodu chceme nájsť všetkých súrodencov Ľudovíta XIV., v policajnom registri chceme nájsť osobu s otlačkami prstov zhodnými s otlačkami z miesta činu, v slovníku chceme nájsť preklad slova.

Algoritmus, ktorý použijeme pri vyhľadávaní závisí hlavne od typu dátovej štruktúru, v ktorej mám určený prvok vyhľadať.

Pri vyhľadávaní prvkov s danou vlastnosťou v lineárnej dátovej štruktúre (postupnosti) môžem použiť tieto základné algoritmy: 

1. lineárny výber - postupne skontrolujem všetky prvky od prvého po posledný, skončím pri poslednom prvku, alebo aj skôr, keď nájdem prvok s požadovanou vlastnosťou: 

<nastav sa na prvý prvok>;

while <nie si na konci postupnosti>and<nie je to hľadaný prvok>

do <posuň sa na ďalší prvok>

Koľko základných operácií (porovnaní) musíme pri tomto postupe vykonať v najhoršom prípade (t.zn. keď musíme prejsť celú postupnosť)? Skúste tento postup naprogramovať pomocou cyklu repeat. 

2. binárny výber - pri tomto postupe využijeme, že postupnosť je usporiadaná vzostupne (od najmenšieho po najväčší) podľa danej vlastnosti. Vždy si vyberieme najskôr stredný prvok z úseku, v ktorom hľadáme. Zistíme, či to nie je ten, ktorý hľadáme. Ak nie je a náš hľadaný prvok je od stredného menší, vyberieme si z poľa len jeho ľavú polovicu (t.j. prvky vľavo od stredného) a opakujeme rovnaký postup, kým nenájdeme hľadaný prvok. Ak bol náš prvok väčší od stredného postupujeme ďalej v pravej polovici. Ak sa hľadaný prvok v postupnosti vôbec nenachádza, bude sa úsek, v ktorom hľadáme stále zmenšovať na polovicu, až zostane prázdny (nulovej dĺžky): 

<úsek, v ktorom hľadám je najskôr celá postupnosť>;

repeat
  <vyber stredný prvok úseku>;

  if <hľadaný prvok> < <stredný prvok>

  then <vezmi ľavý úsek>

  else if <hľadaný prvok> > <stredný prvok>

       then <vezmi pravý úsek>

until (<hľadaný prvok>=<stredný prvok>) or <zostal prázdny úsek>

Koľko porovnaní musíme vykonať v najhoršom prípade? 

Príklad:
Dané je prirodzené číslo N a N-prvková postupnosť celých čísel usporiadaná vzostupne. Zistite pozíciu danej hodnoty H v postupnosti, príp. vypíšte oznam, že hodnota sa v postupnosti nenachádza. Použite lineárny aj binárny výber a pre oba postupy vypíšte, koľko porovnaní sa muselo vykonať pri vyhľadávaní. 

<načítaj N>;

<načítaj usporiadanú postupnosť>;

<načítaj hodnotu H>;

<vyhľadaj pozíciu hodnoty H v postupnosti lineárne>;

writeln(H,'sa nachádza na ',I,' pozícii.');

writeln('Pri lineárnom postupe som vykonal ',POCL,' porovnaní.');

<vyhľadaj pozíciu hodnoty H v postupnosti binárne>;

writeln(H,'sa nachádza na ',J,' pozícii.');

writeln('Pri binárnom postupe som vykonal ',POCB,' porovnaní.');

Rozšírte program tak, že sa nebude zadávať jedna hodnota H na vyhľadanie, ale viac za sebou, pričom posledná bude nula.

1.36. Textový súbor

Súbor je postupnosť dát uložená na disku pod vlastným menom, napr.: list.txt. Úplné meno súboru obsahuje aj názov disku, na ktorom je uložený, aj s adresárovou cestou, napr.:

 c:\dokumenty\posta\list.txt.
Textový súbor obsahuje dáta vo forme textu, t.zn. znaky, usporiadané v riadkoch rôznej dĺžky. Veľkosť súboru je tiež rôzna, závisí od počtu znakov, ktoré obsahuje. Textový súbor si môžeme predstaviť ako list papiera, ktorý je ohraničený len zľava a zhora. Môžeme naň písať riadky ľubovoľnej dĺžky, počet riadkov je tiež ľubovoľný, obmedzený len kapacitou disku. 

Takýto textový súbor si môžeme prezerať a editovať v nejakom textovom editore, napr. Poznámkový blok (Notepad), Windows Commander alebo Volkov Commander (služba F3 View, F4 Edit), textový editor v nejakom programovacom jazyku (napr. Borland Pascal),..

Na konci každého riadku je v súbore neviditeľný znak konca riadku. Môžeme ho znázorniť ako . Textový editor nám ho nezobrazuje, ale v súbore je zapísaný ako dva znaky (#13#10), ktoré označujú koniec riadku. Na konci celého súboru je tiež neviditeľný znak konca súboru (#26). 

S textovým súborom môžeme pracovať aj v programe. Výsledky programu (výstup) nemusíme vypisovať len na obrazovku, ale môžeme ich zapísať do súboru, kde zostanú trvalo uchované. Podobne vstup programu nemusíme realizovať len z klávesnice, ale vstupné dáta môžeme načítať z textového súboru. Využijeme to napríklad vtedy, keď je vstupných dát väčšie množstvo. Pri ladení program spúšťame niekoľkokrát za sebou a je dosť pracné opakovane zadávať množstvo vstupných dát.

Ak chceme v programe pracovať s textovým súborom, musíme si zadeklarovať premennú, ktorá ho bude reprezentovať.

Deklarácia premennej typu textový súbor: 
var <premenná>:text              napr.: var T:text

Príkazy pre prácu s textovým súborom: 

Assign(<premenná typu textový súbor>,<meno súboru>)
napr.: assign(T, 'c:\dokumenty\posta\list.txt')

- týmto príkazom pridelíme premennej typu textový súbor konkrétny súbor z disku. Meno súboru je reťazec. Tento súbor bude premenná reprezentovať vo všetkých ostatných príkazoch. 

Rewrite(<premenná typu textový súbor>)   napr.: Rewrite(T)

- otvorenie súboru pre zápis. Tento príkaz vytvorí nový prázdny súbor na disku, do ktorého môžeme následne zapisovať. Ak daný súbor predtým existoval, jeho obsah sa zmaže. 

Append(<premenná typu textový súbor>)   napr.: Append(T)

- otvorenie súboru pre zápis. Tento príkaz môžeme použiť len pre existujúci súbor, do ktorého môžeme následne zapisovať na koniec. Ak daný súbor predtým neexistoval, objaví sa chybové hlásenie. 

Reset(<premenná typu textový súbor>)   napr.: Reset(T)

- otvorenie súboru pre čítanie. Tento príkaz môžeme použiť len pre existujúci súbor, z ktorého môžeme následne čítať. Ak daný súbor predtým neexistoval, objaví sa chybové hlásenie. 

write/writeln(<premenná typu textový súbor>,<zoznam zapisovaných údajov>)
napr.: write(T,'Pocet deti je ', CH+D,'.')

- zápis údajov do textového súboru. Údaje môžu byť typu znak, reťazec, číslo a oddeľujeme ich čiarkami. Príkazy používame rovnako ako pri výpise na obrazovku s tým rozdielom, že prvý parameter príkazu je premenná typu textový súbor. 

read/readln(<premenná typu textový súbor>,<zoznam premenných>)
napr.: readln(T,A,B,C)

- čítanie údajov z textového súboru. Premenné môžu byť typu znak, reťazec, číslo a oddeľujeme ich čiarkami. Príkazy používame rovnako ako pri čítaní z klávesnice s tým rozdielom, že prvý parameter príkazu je premenná typu textový súbor. 

close(<premenná typu textový súbor>     napr.: close(T)

- zatvorenie textového súboru. Na konci práce so súborom, musíme súbor zatvoriť. Ak do súboru zapisujeme a zabudneme ho na konci programu zatvoriť, pravdepodobne v súbore budú chýbať posledné údaje, ktoré sme doň zapisovali. Je to z toho dôvodu, že príkaz write/writeln nezapisuje dáta do súboru okamžite, dáta sa do súboru prenášajú po väčších blokoch. Príkazom close sa prenesie na disk aj posledný blok dát. 

Funkcie pre prácu s textovým súborom: 

eoln(<premenná typu textový súbor>)    napr.: eoln(T)

- logická funckia. Vráti TRUE/FALSE, ak smerník pre čítanie je/nie je na konci aktuálneho riadku súboru. 

eof(<premenná typu textový súbor>)    napr.: eof(T)

- logická funckia. Vráti TRUE/FALSE, ak smerník pre čítanie je/nie je na konci súboru. 

OBE FUNKCIE SA POUŽÍVAJÚ LEN PRI ČÍTANÍ.


Príklad:
Daný je počet prvkov N a N-prvková postupnosť prirodzených čísel. Zistite a vypíšte počet výskytov jednotlivých čísel v postupnosti.

Nasledujúce dva programy riešia danú úlohu, pričom prvý realizuje vstup z klávesnice a výstup na obrazovku, druhý program číta vstupné dáta zo súboru DATA.IN a výsledky zapisuje do súboru DATA.OUT. 

	type TPOLE=array[1..100]of integer;

var A:TPOLE;

    N,POCET,I,J:integer;

begin

  write('Zadaj pocet prvkov: '); readln(N);

  writeln('Zadaj hodnoty prvkov: ');

  for I:=1 to N

  do read(A[I]);

  for I:=1 to N

  do if A[I]>0

     then begin

           POCET:=0;

           for J:=I to N

           do if .....

              then POCET:=POCET+1;

           writeln(A[I],' sa vyskytuje ',POCET,'-krat.');

           A[I]:=.....

          end

end.
	type TPOLE=array[1..100]of integer;

var A:TPOLE;

    N,POCET,I,J:integer;

    VST,VYS:text;

begin

  assign(VST,'DATA.IN'); assign(VYS,'DATA.OUT');

  reset(VST); rewrite(VYS);

  readln(VST, N);

  for I:=1 to N

  do read(VST, A[I]);

  for I:=1 to N

  do if A[I]>0

     then begin

           POCET:=0;

           for J:=I to N

           do if .....

              then POCET:=POCET+1;

           writeln(VYS, A[I],' sa vyskytuje ',POCET,'-krat.');

           A[I]:=.....

          end;

close(VST); close(VYS)

end.



Doplňte v programe chýbajúce časti. Koľko porovnaní sa vykoná pre N-prvkovú postupnosť v najhoršom prípade? 

	Vstupný súbor by mohol vyzerať takto: 

10
1  7  1  1  10  7  10  7  7  3 3 


	Výstupný súbor potom bude: 

1 sa vyskytuje 3-krat.
7 sa vyskytuje 4-krat.
10 sa vyskytuje 2-krat.
3 sa vyskytuje 1-krat.



Prečo sa hodnota 3 vyskytuje v danej postupnost len 1-krat?

1.36.1. Chybové hlásenia pri práci s textovým súborom

File not found 

Pri otváraní (Reset, Append) nenašiel daný súbor. Buď súbor neexistuje, alebo je zle zadaná adresárová cesta k súboru. 

Path not found 

Pri otváraní súboru bola zle zadaná adresárová cesta, daný adresár neexistuje. 

File not open for input 

Súbor nie je otvorený pre čítanie. Súbor sme otvorili príkazom rewrite/append a chceme z neho čítať. 

File not open for output 

Súbor nie otvorený pre zápis. Súbor sme otvorili príkazom reset a chceme doň zapisovať, alebo použiť funkciu eoln/eof. 

Príklady: 

1. Daný je textový súbor TEXT.IN. Zistite, koľko obsahuje a) znakov, b) riadkov, c) viet (veta je ukončená bodkou, výkričníkom, alebo otáznikom). 

2. Daný je textový súbor MAJETOK.IN, obsahujúci majetkové priznanie obsahujúce niekoľko položiek. Každá položka je zapísaná v jednom riadku v tvare: názov majetku, medzera, hodnota v Sk. Zistite celkovú hodnotu majetku. 


3. Daný je textový súbor CISLA.IN obsahujúci postupnosť čísel. Vytvorte textový súbor HVIEZDY.OUT, do ktorého zapíšete postupnosť "hviezdičkových" riadkov, kde dĺžky riadkov sú určené číslami z danej postupnosti. 
	CISLA.IN
	HVIEZDY.OUT

	5    
1
2
8
	*****
*
**
********    


4. Milionár: V textovom súbore OTAZKY.IN sú zapísané znenia otázok oddelené prázdnymi riadkami. Za textom každej otázky nasledujú 4 možnosti odpovedí (A,B,C,D). V textovom súbore ODPOVEDE.IN je zapísaná postupnosť znakov, predstavujúcich správne odpovede k otázkam v prvom súbore. Vytvorte program, ktorý bude postupne po jednej otázke čítať zo súboru, vypisovať ju na obrazovku a čítať z klávesnice užívateľove odpovede. Za správnu odpoveď získa užívateľ bod.
Doplnok: užívateľ môže pri každej otázke použiť žolíka '50:50', pri odpovedi bez žolíka sa mu započítajú 2 body, so žolíkom 1 bod. 
	OTAZKY.IN 
	ODPOVEDE.IN

	Ako sa volá hlavné mesto Talianska?

A. Talian    B. Rím

 C. Vatikán  D. Pelikán

Čo je to tritical?

 A. chemická zlúčenina  B. hmyz

C. praveký jašter         D. obilnina
	B   

D


1.37. Generovanie náhodných údajov

Vygenerovať náhodný údaj z nejakej množiny znamená náhodne vybrať jednu hodnotu z danej množiny. V jazyku Pascal máme tieto nástroje na generovanie náhodných čísel: 

random(<celé číslo>) 

- funkcia, ktorá vráti náhodné celé číslo z intervalu <0,<celé číslo>-1>. Napr. random(10) vráti celé číslo z intervalu <0,9>. Celé číslo, ktoré je parametrom funkcie random, predstavuje "dĺžku" intervalu (počet celých čísel), z ktorého vyberá náhodné číslo, pričom dolná hranica intervalu je vždy 0. 

randomize 

- procedúra, ktorú stačí použiť len raz na začiatku programu, alebo pred prvým použitím funkcie random. Je to vlastne "naštartovanie" generátora náhodných čísel a spôsobí, že pri každom ďalšom spustení programu bude generovať iné čísla. 

random 

- funkcia použitá bez parametra vráti náhodné reálne číslo z intervalu <0,1) 

Príklad použitia:

Užívateľ zadá prirodzené číslo N predstavujúce počet prvkov postupnosti náhodných čísel, ktoré má počítač vygenerovať. Program sa spýta užívateľa, či chce postupnosť celých alebo reálnych čísel a vygeneruje ju. 

uses Crt;

var N:integer;

    VYBER:char;

  procedure CELE;

  var I:integer;

  begin

    for I:=1 to N

    do write(random(100),' ')

  end;

  procedure REALNE;

  var I:integer;

  begin

    for I:=1 to N

    do writeln(random:0:10)

  end;

BEGIN

  write('Zadaj, kolko cisel chces: '); readln(N);

  writeln('Zadaj, ake chces cisla:');

  writeln('a) cele z intervalu <0,99>');

  writeln('b) realne z intervalu <0,1)');

  VYBER:=upcase(readkey);

  case VYBER of

    'A':CELE;

    'B':REALNE;

  end;

  readln

END.

Úloha pre vás: rozšírte ponuku programu o možnosti generovať

- celé čísla z intervalu <100,150>
- reálne čísla z intervalu <0,10)
- reálne čísla z intervalu <5,15)
- znaky z intervalu 'a'..'z'
- logické hodnoty
- farby z množiny {biela, žltá, oranžová, červená, zelená, modrá, fialová}

Ak chceme vygenerovať náhodné celé číslo z intervalu <A,B> a uložiť ho do premennej C:integer, použijeme príkaz

C:=......................................... (doplňte).

Ak chceme vygenerovať náhodné reálne číslo z intervalu <A,B) a uložiť ho do premennej C:real, použijeme príkaz

C:=......................................... (doplňte).

Ak chceme vygenerovať náhodný nečíselný údaj, spravíme to tak, že ...............................

Príklady: 

1. Vytvorte program, ktorý náhodne "rozhadzuje" farebné obdĺžničky po obrazovke, t.j. opakovane generuje náhodné súradnice X, Y a náhodnú farbu a vykresľuje políčko s určenými súradnicami určenou farbou. Program končí, ak užívateľ stlačí ľubovoľnú klávesu. 

2. Do stredu obrazovky vypisujte v určitých časových intervaloch náhodné čísla z intervalu <0,9> (stále na rovnaké políčko), pokiaľ užívateľ nestlačí ľubovoľnú klávesu. (Na obrazovke sa točia čísla podobne, ako keď sa žrebuje.)

3. Doplnok: Úlohou užívateľa je zastaviť generovanie, keď je na obrazovke číslo 0. Ak bol úspešný, započíta sa mu bod. Po 10 pokusoch sa vyhodnotí jeho úspešnosť. 

4. Program vygeneruje dve celé čísla - operandy a aritmetickú operáciu (+ - * /). Vypíše na obrazovku vygenerovaný výraz v tvare <operand><operátor><operand>= a od užívateľa načíta číslo predstavujúce výsledok výrazu. Ak je užívateľov výsledok správny, započíta sa mu bod. Po 10 príkladoch program vyhodnotí úspešnosť užívateľa a oznámkuje ho. 

1.38. Štandardné procedúry a funkcie pre typ STRING

length(<reťazec>) 

- funkcia, ktorá vráti celé číslo predstavujúce dĺžku reťazca 

pos(<podreťazec>,<reťazec>) 

- funkcia vráti pozíciu prvého výskytu podreťazca v reťazci, príp. nulu, ak sa podreťazec v reťazci nenachádza 

copy(<reťazec>,<pozícia>,<počet znakov>) 

- funkcia vráti podreťazec, ktorý vznikne výberom daného počtu znakov z daného reťazca od danej pozície 

insert(<podreťazec>,<reťazec>,<pozícia>) 

- vloží do reťazca daný podreťazec na danú pozíciu 

delete(<reťazec>,<pozícia>,<počet znakov>) 

- v danom reťazci zmaže od danej pozície daný počet znakov 

str(<číslo>,<reťazec>) 

- procedúra zmení číslo na reťazec 

val(<reťazec>,<číslo>,<kód chyby>) 

- procedúra zmení reťazec na číslo a v kóde chyby vráti nulu; v prípade že reťazec nepredstavuje zápis čísla, bude výsledné číslo nula a v kóde chyby bude uložená pozícia prvého chybného znaku v reťazci 

Príklady použitia: 

var I,K:integer;

    C:real;

    S,R:string

	Stav premenných pred
	Príkaz
	Stav premenných po
	Poznámka

	S='dva'
	I:=length(S)
	I=3
	length(S) je to isté ako ord(S[0])

	S='vodovod'
	I:=pos('od',S)
	I=2
	

	S='BYSTROZRAKÝ'
	I:=pos('rak',S)
	I=0
	podreťazec nenašiel kvôli rozdielu medzi malými a veľkými písmenami

	S='straka'
	R:=copy(S,2,4)
	R='trak'
	

	S='konár'
	insert('mi',S,3)
	S='kominár'
	

	S='slovák'
	delete(S,3,2)
	S='slák'
	

	I=72
	str(I,S)
	S='72'
	

	C=2.34
	str(C:6:3,S)
	S=' 2.340'
	čísla premieňané na reťazec môžeme formátovať ako pri výpise na obrazovku

	S='-45.6'
	val(S,C,K)
	C=-45.6, K=0
	reťazec sa podarilo premeniť na číslo

	S='-45.6'
	val(S,I,K)
	I=0, K=4
	reťazec nemožno premeniť na celé číslo kvôli 4. znaku

	S='-45'
	val(S,I,K)
	I=-45, K=0
	reťazec sa podarilo premeniť na číslo


1.38.1. Príklady

(v zátvorkách sú uvedené funkcie a procedúry, ktoré by ste mohli v príklade využiť) 

1. Vytvorte program, ktorý načíta reťazec, vykoná s ním nasledovné úpravy a po každej ho vypíše:

- každú čiarku zmení na bodkočiarku
- každú za sebou idúcu dvojicu znakov 'xx' zmení na '2x'
- každý začiatok slova (slová sú oddelené medzerami, t.zn. pred každým slovom s výnimkou prvého je medzera) zmení na veľké písmeno.
(length) 

2. Z daného reťazca S odstráňte všetky medzery a vypíšte ho. (length, delete) 

3. Do daného reťazca vložte za každý znak medzeru a vypíšte ho. (length, insert) 

4. V danom reťazci S je uložené meno a priezvisko oddelené medzerou. Do reťazca M uložte zvlášť meno a do reťazca P priezvisko. Napr. z reťazca S='Kubo Zajac' vytvorte reťazce M='Kubo' a P='Zajac'. (pos, copy, length) 

5. Vyriešte bez požitia typu real: Načítajte reálne číslo (do reťazcovej premennej), zistite a vypíšte, koľko má desatinných miest. (pos, length)
Vypíšte jeho desaťnásobok. (delete, insert) 

6. Vygenerujte náhodný reťazec zložený z písmen abecedy náhodnej dĺžky max. 80. 

7. Vygenerujte náhodný text a zapíšte ho do súboru - počet riadkov bude náhodné číslo z intervalu 1..20, dĺžka každého riadku bude náhodné číslo z intervalu 1..50, v riadkoch budú náhodné abecedné znaky. 

8. Bez použitia počítača určte výstup programu: 

var S,R:string;

begin

  S:='REKLAMA';

  R:=copy(S,3,5);

  S:=S+R;

  insert(' ',S,8);

  S[13]:='E';

  S[14]:='!';

  writeln(S)

end.

9. V premennej S:string je uložený reťazec. Zapíšte príkazy, ktorými:
- zmažete posledné 3 znaky 

- prvý znak presuniete na koniec 

– do stredu vložíte znak '-' 

1.39. Procedúry s parametrami

Príklad č.1: Daný je počet prvkov N a N-prvková postupnosť celých čísel A[1..N]. Vytvorte nové pole B, uložte doňho všetky hodnoty z poľa A tak, aby sa každá vyskytovala len raz a ich počet uložte do premennej P. Obe polia vypíšte.

Napr.: ak N=7 a A=(2,1,5,1,5,5,4), potom P=4 a B=(2,1,5,4).

Riešenie: Úlohu si rozdelíme na podúlohy, ktoré budú riešiť tieto podprogramy: CITAJ_A (načíta N a pole A), VYROB_B (naplní pole B a premennú P), VYPIS_A (vypíše N-prvkové pole A) a VYPIS_A (vypíše P-prvkové pole B). Hlavný program potom bude vyzerať takto: 

begin

  CITAJ_A;

  VYROB_B;

  VYPIS_A;

  VYPIS_B

end.

Posledné dve procedúry budú vyzerať takto: 

	procedure VYPIS_A;

var I:integer;

begin for I:=1 to N
      do write(A[I]:5);       

      writeln

end
	procedure VYPIS_B;

var I:integer;

begin for I:=1 to P
      do write(B[I]:5);

      writeln

end


Keď obe procedúry porovnáme, zistíme, že sa líšia len názvami premenných. Obe vyjadrujú ten istý postup spracovania, ale každá pracuje s inými hodnotami. V tomto prípade je výhodnejšie zadeklarovať miesto týchto dvoch procedúr jednu "univerzálnu", ktorá bude pracovať s "neznámym" poľom X a neznámou premennou Y vyjadrujúcou počet prvkov. Za "neznáme" X,Y dosadíme pri volaní procedúry premenné A,N a potom B,P. 

Deklarácia bude vyzerať takto: 
procedure VYPIS(X:POLE;Y:integer);

var I:integer;

begin for I:=1 to Y
      do write(X[I]:5);

      writeln

end

Volania procedúry v hlavnom programe budú vyzerať takto: 

begin

  CITAJ_A;

  VYROB_B;

  VYPIS(A,N);

  VYPIS(B,P)

end.

Úloha pre vás: Vypracujte ostatné deklarácie.

1.39.1. Parameter

je premenná cez ktorú sa odovzdávajú vstupné a výstupné hodnoty podprogramu. 

Parametre rozdeľujeme na: 

· formálne - uvedené sú v hlavičke deklarovaného podprogramu. Používajú sa ako lokálne premenné - vznikajú pri volaní podprogramu, zanikajú po jeho skončení, mimo podprogramu sú nepoužiteľné. Slúžia na uloženie hodnôt, s ktorými bude podprogram pracovať. 

· skutočné - uvedené sú pri volaní podprogramu, sú to skutočné hodnoty, s ktorými bude podprogram pracovať. 

Deklarácia podprogramu s parametrami: 

procedure <meno> (<zoznam parametrov a ich typov oddelených bodkočiarkami>);

<lokálne deklarácie>;

begin

  <príkazy>

end

Volanie podprogramu s parametrami: 

<meno>(<zoznam parametrov oddelených čiarkami>)

POZOR! 

· typ parametra v hlavičke podprogramu musí byť jeden identifikátor. Napr. nesmieme zapísať procedure XYZ(A:array[1..10]of char), ale musíme to zapísať: procedure XYZ(A:POLE), kde type POLE=array[1..10]of char 

· skutočné a formálne parametre sa musia zhodovať v počte, poradí a type. 

Príklad č.2: Zadeklarujte procedúru s parametrami PIS_ZNAKY(C:char;P:integer), ktorá vypíše za sebou do riadku P znakov C a kurzor nechá za posledným vypísaným znakom. Vytvorte najskôr skúšobný hlavný program, ktorým odskúšate jej funkčnosť s volaniami napr. PIS_ZNAKY('@',5); PIS_ZNAKY('.',10).

Potom vytvorte nový hlavný program, ktorý načíta prirodzené číslo N a s pomocou tejto procedúry vypíše na obrazovku tieto útvary s rozmerom N (nepoužívajte gotoxy!): 

****!   ****        *******

***!!    ****        *****

**!!!     ****        ***

*!!!!      ****        *

Parametre rozdeľujeme na: 

· vstupné (volané hodnotou) - do podprogramu sa odovzdáva len hodnota premennej, s ktorou sa pracuje. Po skončení podprogramu hodnota premennej zostáva nezmenená. 

· vstupno-výstupné (volané adresou, referenciou) - do podprogramu sa odovzdáva celá premenná, podprogram s ňou pracuje a môže ju zmeniť. V hlavičke podprogramu sú odlíšené slovkom var. Napríklad: 

· var X,Y:integer;

·     procedure SKUSKA(A:integer; var B:integer);

·     begin  A:=A+2;

·            B:=B+A

·     end;
· BEGIN  X:=5; Y:=8;

·        SKUSKA(X,Y);

·        writeln(X,' ',Y)

· END.

Pri volaní podprogramu sa hodnota skutočného parametra X vloží do formálneho parametra A. Keďže je to len vstupný parameter, po skončení podprogramu sa jeho nová hodnota neposiela naspäť. Ale, parameter B dostane na začiatku vstupnú hodnotu Y a na konci pošle naspäť premennej Y novú (výstupnú) hodnotu. Čo nám tento program vypíše? Čo vypíše tento program, ak volanie SKUSKA(X,Y) nahradíme volaním a) SKUSKA(3,Y)   b) SKUSKA(X+3,Y)   c) SKUSKA(X,3)   d) SKUSKA(X,Y+3)?
 
Využitie parametrov: 

· ak sa v programe opakuje tá istá postupnosť príkazov s rôznymi premennými 

· ak pri deklarácii podprogramu ešte neviem názov premennej s ktorou bude program pracovať 

Príklady: 

1. Program načíta prirodzené čísla M,N, potom M známok žiaka A a N známok žiaka B. Vypíše, ktorý z nich má lepší priemer. 

2. Zadeklarujte procedúru NSD(A,B:integer; var C:integer), ktorá pre dané čísla A,B nájde ich najväčšieho spoločného deliteľa C pomocou Euklidovho algoritmu. 

3. Do predchádzajúceho programu zadeklarujte procedúru NSN(A,B:integer; var C:integer), ktorá pre dané čísla A,B nájde ich najmenší spoločný násobok C, tak, že ich súčin vydelí ich najväčším spoločným deliteľom. 

4. Program načíta zlomok - 2 celé čísla predstavujúce čitateľa a menovateľa. Zlomok upraví na základný tvar a výsledok vypíše. Využite vyššie uvedené procedúry. 

Funkcia

Funkcia je podprogram, ktorý vráti jednu výstupnú hodnotu cez svoje meno. Môže byť s parametrami aj bez parametrov. 

Deklarácia: 

function <meno> (<parametre>):<typ výstupnej hodnoty>;

<lokálne deklarácie>;

begin

  <príkazy>

end  

Parametre funkcie sa deklarujú aj používajú rovnako ako pri procedúre. Typ výstupnej hodnoty môže byť jednoduchý (číselný, znakový, logický, vymenovaný, interval) alebo string. Nesmie to byť štruktúrovaný typ. V príkazovej časti sa do mena funkcie priradí výstupná hodnota.

Príklad: 

function VACSI(A,B:integer):integer;

begin

  if A>B

  then VACSI:=A

  else VACSI:=B

end

Volanie funkcie: 

<meno>(<parametre>)

Meno funkcie je nositeľom hodnoty, preto funkciu používame ako hodnotu príslušného typu (môžeme ju niekam priradiť, vypísať, použiť vo výraze) napr.:

X:=VACSI(3,5); Y:=VACSI(X-4,4); Z:=VACSI(Y,VACSI(X,2));

writeln(Z)

Príklady.

1.  Zadeklarujte procedúru s vhodnými parametrami na výpočet N-faktoriálu N!=1.2.3....N. Vytvorte hlavný program, ktorý s pomocou podprogramu vypočíta pre dané prirodzené čísla N, K výraz: N!/K!/(N-K)! Vyriešte tento istý príklad s použitím funkcie a oba programy porovnajte. Čo vypíše posledný príkaz?

2. Zadeklarujte funkciu s vhodnými parametrami, ktorá zistí, či sa dané dva reťazce rýmujú (zhodujú dvojznakovou koncovkou). 

3. Zadeklarujte funkciu, ktorá premení teplotu danú v °C na jednotky Fahrenheita podľa vzťahu F=9/5*C+32. Vytvorte hlavný program, ktorý načíta prirodzené číslo D a vypíše 20-riadkovú tabuľku, v ktorej prvom stĺpci sú teploty v °C počnúc od nuly s krokom D a v druhom stĺpci príslušné teploty v stupňoch Fahrenheita. (Ak bude D=5, v prvom stĺpci budú čísla 0,5,10,15,...) 

4. Zadeklarujte funkciu, ktorá v daných dvoch reťazcoch nájde ich najdlhší spoločný začiatočný podreťazec, napr. SP_ZAC('pero','perina')='per' 

5. Daný je reťazec DATUM predstavujúci dátum v tvare 'deň.mesiac.rok' . Deň i mesiac môžu byť jedno- i dvojciferné. Vytvorte funkcie: * DEN(DATUM:string):integer ... z daného dátumu vráti číslo dňa * MESIAC(DATUM:string):integer ... z daného dátumu vráti číslo mesiaca* ROK(DATUM:string):integer ... z daného dátumu vráti číslo roka

1.40. Triedenie

Ak potrebujeme často vyhľadávať prvky v skupine dát podľa nejakej vlastnosti, je dobré mať skupinu dát usporiadanú podľa danej vlastnosti. Napr. ak čitateľ v knižnici chce nájsť knižku od nejakého spisovateľa, využije na vyhľadanie menný katalóg, kde sú usporiadané záznamy o knihách podľa mien autorov. Ak chce nájsť knižku podľa názvu, využije názvový katalóg, kde sú záznamy usporiadané podľa názvov kníh. Vyhľadávanie v usporiadanej postupnosti je neporovnateľne rýchlejšie ako v neusporiadanej.

Postupnosť a1, a2, a3, ..., an môže byť usporiadaná: 

· vzostupne (od najmenšieho po najväčší prvok): a1 <= a2 <= a3 <= ... <= an 

· zostupne (od najväčšieho po najmenší prvok): a1 >= a2 >= a3 >= ... >= an 

Napr. postupnosť čísel 2,8,5,2,9,1,8 môžem usporiadať vzostupne: 1,2,2,5,8,8,9 alebo zostupne: 9,8,8,5,2,2,1. Usporiadaním postupnosti počet prvkov zostáva nezmenený, takisto hodnoty prvkov sa nemenia, len majú iné poradie.

Úloha: Vytvorte program, ktorý danú postupnosť čísel usporiada vzostupne. Postupnosť je daná počtom prvkov a hodnotami jednotlivých prvkov. Sformulujeme najskôr vstupnú a výstupnú podmienku, t.zn. presne určíme, čo je dané a čo chceme dostať:

{VST: Dané je prirodzené číslo N a N-prvková postupnosť celých čísel a1, a2, a3, ..., an}

{VÝS: Usporiadaná postupnosť tých istých čísel, kde platí: a1 <= a2 <= a3 <= ... <= an} 

Keďže hodnoty prvkov postupnosti sa nesmú zmeniť, jediné operácie, ktoré s nimi budeme môcť robiť sú porovnanie (menší, väčší) a vzájomné výmeny. Zopakujme si algoritmus výmeny obsahu dvoch premenných X a Y. Použijeme tretiu pomocnú premennú Z, rovnakého typu ako premenné X,Y. Výmenu uskutočníme tromi priradeniami: 

Z:=X; X:=Y; Y:=Z

Existujú viaceré typy algoritmov na vyriešenie tejto úlohy, my sa zoznámime s dvoma najjednoduchšími. 

1.40.1. Triedenie cez maximum - MAXSORT

Postup: Nájdem najväčší prvok v postupnosti A[1..N]. Ten patrí na posledné miesto, preto ho vymením s posledným prvkom. Tento prvok je už tam, kde má byť, preto si ho už ďalej nebudem všímať. Zopakujem tento postup pre "skrátenú" postupnosť A[1..N-1]. Ďalším opakovaním nakoniec dostanem usporiadanú postupnosť. Hrubá schéma algoritmu: 

for I:=N downto .....

do begin <nájdi maximum v úseku postupnosti A[1..  ]>;

         <vymeň ho s prvkom A[  ]>

   end

Doplňte chýbajúce časti. Pri triedení nám však nestačí nájsť len hodnotu maxima, ale kvôli následnej výmene prvkov v postupnosti potrebujeme poznať aj jeho pozíciu - index. Preto pri jeho nájdení musíme odložiť do nejakej pomocnej premennej aj jeho index. 

1.40.2. Triedenie prebublávaním - BUBBLESORT

Keďže chceme postupnosť čísel vzostupne usporiadať, budeme sa snažiť dostať menšie prvky vľavo a väčšie vpravo. Princíp prebublávania spočíva v porovnávaní dvojíc susedných prvkov a ich následnej výmene: 

if A[I] > A[I+1]

     then <vymeň prvky A[I], A[I+1]>

Samozrejme, takýchto porovnaní a prípadných výmen musíme urobiť viac. Jednoznačne musíme porovnať všetky dvojice susedných prvkov - najlepšie systematicky zľava doprava: 

for I:=1 to .....

do if A[I] > A[I+1]

      then <vymeň prvky A[I], A[I+1]>

Úloha: Doplňte hornú hranicu cyklu for.

Napíš, ako bude vyzerať postupnosť 2,8,5,2,9,1,8 po aplikácii horeuvedeného algoritmu: ............................. Čo treba do algoritmu doplniť, aby triedil celú N-prvkovú postupnosť? Navrhni vylepšenia algoritmu, aby sa počítač pri triedení postupnosti čo najmenej "narobil".
Porovnaj obe metódy triedenia.


Príklad: 
Vytvorte ponukový program, ktorý na začiatku vygeneruje náhodnú postupnosť čísel z intervalu <1,20>. (Pred generovaním si vždy vypýta počet prvkov postupnosti.) Potom opakovane realizuje nasledovnú ponuku: 

1. Výpis aktuálnej postupnosti 

2. Vygenerovanie novej postupnosti 

3. Vzostupné utriedenie postupnosti MAXSORTom 

4. Vzostupné utriedenie postupnosti BUBBLESORTom 

5. Vzostupné utriedenie postupnosti MINSORTom 

6. Zostupné utriedenie postupnosti ľubovoľným spôsobom 

7. Koniec 

MINSORT je triedenie cez minimum, ktoré je v princípe rovnaké, ako triedenie cez maximum. Opakovane vyhľadávame najmenší prvok v úseku a vymieňame ho so začiatočným prvkom. Počas behu programu nezotierajte obrazovku, aby bolo možné porovnať pôvodnú postupnosť s usporiadanou. Skúste vymyslieť vlastný algoritmus triedenia. 

